10. Calculus: Derivatives, Series, Limits, Integration

2016-07-08

```
$Version
10.0 for Mac OS X x86 (64-bit) (December 4, 2014)
```

10.1 Derivatives

10.1.1 Common and Partial Derivatives

`D[f,x]`	Derive **f** once w.r.t. **x**
`D[f, x, y]`	Derive **f** w.r.t. **x**, then w.r.t. **y**
`D[f, {x, nx}]`	Derive **f** w.r.t. **x** **nx** times
`D[f, {x, nx} {y, ny}, ...}`	Derive **f** w.r.t. **x** **nx** times, then w.r.t. **y** **ny** times, etc.

```
f = Sin[a x]
Sin[a x]

g = D[f, x]
a Cos[a x]

h = D[f, {x, 4} ]
```

$a^4 \mathrm{Sin}[a\,x]$

```
f = Exp[ a x + b y + c z ]
```

$e^{a x+b y+c z}$

```
g = D[f, {x, 2}, {y, 3}, {z, 2} ]
```

$a^2 b^3 c^2 e^{a x+b y+c z}$

```
Clear[f]
g = D[f, x]
0

D[Log[x]^2, x]
```

$\frac{2\,\mathrm{Log}[x]}{x}$

Legendrepolynome

Legendre polynomials may be generated from the generating function:

$$\frac{1}{\sqrt{1-2\,x\,\alpha+\alpha^2}} = \sum_{n=0}^{\infty} \alpha^n P_n[x], \quad \left[\frac{d^k}{d\alpha^k}\frac{1}{\sqrt{1-2\,x\,\alpha+\alpha^2}}\right]_{\alpha\to 0} = \left[\sum_{k=0}^{\infty} k!\ \alpha^k P_k[x]\right]_{\alpha\to 0}$$

```
legpoly[n_, x_] := D[1/Sqrt[1 - 2 a x + a^2], {a, n}]/n! /.
a -> 0

legpoly[2, x]
```

$\frac{1}{2}\left(-1+3\,x^2\right)$

```
f = legpoly[10, x] // Apart
```

$$-\frac{63}{256}+\frac{3465 x^2}{256}-\frac{15015 x^4}{128}+\frac{45045 x^6}{128}-\frac{109395 x^8}{256}+\frac{46189 x^{10}}{256}$$

```
LegendreP[10, x] // Apart
```

$$-\frac{63}{256}+\frac{3465 x^2}{256}-\frac{15015 x^4}{128}+\frac{45045 x^6}{128}-\frac{109395 x^8}{256}+\frac{46189 x^{10}}{256}$$

```
f /. x -> 1
```

1

Normierung für x=1 -> P=1

```
f = LegendreP[3, Cos[th]]
```

$$\frac{1}{2}\left(-3 \,\mathrm{Cos}[th]+5 \,\mathrm{Cos}[th]^3\right)$$

```
g = Expand[f, Trig -> True]
```

$$\frac{3 \,\mathrm{Cos}[th]}{8}+\frac{5 \,\mathrm{Cos}[th]^3}{8}-\frac{15}{8}\mathrm{Cos}[th]\,\mathrm{Sin}[th]^2$$

```
TrigReduce[g]
```

$$\frac{1}{8}\left(3 \,\mathrm{Cos}[th]+5 \,\mathrm{Cos}[3\, th]\right)$$

10.1.2 Derivatives of Unknown Functions

```
Clear[f]
```

```
D[ f[x]^2, x]
```

$2 f[x] f'[x]$

```
D[ x f[x]^2, x]
```

$f[x]^2 + 2 x f[x] f'[x]$

```
D[%, x]
```

$4 f[x] f'[x] + 2 x f'[x]^2 + 2 x f[x] f''[x]$

For functions of serveral variables the derivatives are declared in a list attached to the upper right of thename.

```
Clear[g];
D[g[x, y], x]
```

$g^{(1,0)}[x, y]$

```
D[g[x, y], y]
```

$g^{(0,1)}[x, y]$

```
D[g[x, y], x, y]
```

$g^{(1,1)}[x, y]$

```
D[g[x^2, y^3], x]
```

$2 x g^{(1,0)}[x^2, y^3]$

```
D[g[x^2, y^3], y]
```

$3\, y^2\, g^{(0,1)}[x^2, y^3]$

```
D[g[x^2, y^3], {x, 2}]
```

$2\, g^{(1,0)}[x^2, y^3] + 4\, x^2\, g^{(2,0)}[x^2, y^3]$

```
% /. x -> 0
```

$2\, g^{(1,0)}[0, y^3]$

10.1.3 Total Derivatives

`Dt[f]`	total differential `df`
`Dt[f,x]`	total derivative `df/dx`
`Dt[f,x,y,...]`	Multiple total derivative `(d/dx)(d/dy)(...) f`
`Dt[f, x, Constants -> {c1, c2, ...} ]`	Total derivative with c_i = `const`.
`z/: Dt[z,x] = 0`	Set `dz/dx = 0`
`SetAttributes[c, Constant]`	define `c` to be a constant in all cases

The operator **D[f, x, y, z, ...]** denotes a derivative where x, y, z are independent of each other.
Whereas in **Dt[f, x, y, z, ...]** all variables which have not been declared constant are asssumed to depend on the variable of the actual derivation.

```
D[x^2 + y^2, x]
```

2 x

```
Dt[x^2 + y^2, x]
```

2 x + 2 y Dt[y, x]

```
% /. Dt[y, x] -> D[y[x], x]
```

2 x + 2 y y′[x]

```
D[x^2 + y[x]^2, x]
```

2 x + 2 y[x] y′[x]

```
y /: Dt[y, x] = 0
```

0

```
Dt[x^2 + y^2 + z^2, x]
```

2 x + 2 z Dt[z, x]

```
Clear[y];
Dt[x^2 + y^2 + z^2, x]
```

2 x + 2 z Dt[z, x]

```
Dt[x^2 + y^2 + z^2, x, Constants -> {z}]
```

2 x + 2 y Dt[y, x, Constants → {z}]

```
SetAttributes[c, Constant]
```

```
Dt[a^2 + c x^2, x]
```

2 c x + 2 a Dt[a, x]

```
Dt[a^2 + c[x] x^2, x]
```

2 x c[x] + 2 a Dt[a, x]

```
Dt[x^2 + c y^2]
```

2 x Dt[x] + 2 c y Dt[y]

```
% /. Dt[y] -> dy
```

2 c dy y + 2 x Dt[x]

The total derivative may also be used to change the dependent variable in a differential equation. For example, in the differential operator y'' + y' + y the dependent variable y(x) is replaced by $z(x)/v(x)^{1/4}$. The resulting differential operator has no longer a first derivative of the dependent variable (= normal form of a second order differential operator):

```
Clear[x, R, V, f0, f1, f2]
```

```
f0 = y = z v^(-1/4)
```

$$\frac{z}{v^{1/4}}$$

```
f1 = Dt[f0, x]
```

$$-\frac{z\,\mathrm{Dt}[v,x]}{4\,v^{5/4}}+\frac{\mathrm{Dt}[z,x]}{v^{1/4}}$$

```
f2 = Dt[f1, x]
```

$$\frac{5\,z\,\mathrm{Dt}[v,x]^2}{16\,v^{9/4}}-\frac{z\,\mathrm{Dt}[v,\{x,2\}]}{4\,v^{5/4}}-\frac{\mathrm{Dt}[v,x]\,\mathrm{Dt}[z,x]}{2\,v^{5/4}}+\frac{\mathrm{Dt}[z,\{x,2\}]}{v^{1/4}}$$

```
ft0 = Dt[v, {x, 2}] / 4 f0
```

$$\frac{z\,\mathrm{Dt}[v,\{x,2\}]}{4\,v^{1/4}}$$

```
ft1 = Dt[v, x] / 2 f1 // Expand
```

$$-\frac{z\,\mathrm{Dt}[v,x]^2}{8\,v^{5/4}}+\frac{\mathrm{Dt}[v,x]\,\mathrm{Dt}[z,x]}{2\,v^{1/4}}$$

```
ft2 = v f2 // Expand
```

$$\frac{5\,z\,\mathrm{Dt}[v,x]^2}{16\,v^{5/4}}-\frac{z\,\mathrm{Dt}[v,\{x,2\}]}{4\,v^{1/4}}-\frac{\mathrm{Dt}[v,x]\,\mathrm{Dt}[z,x]}{2\,v^{1/4}}+v^{3/4}\,\mathrm{Dt}[z,\{x,2\}]$$

```
ft0 + ft1 + ft2
```

$$\frac{3\,z\,\mathrm{Dt}[v,x]^2}{16\,v^{5/4}}+v^{3/4}\,\mathrm{Dt}[z,\{x,2\}]$$

```
(ft0 + ft1 + ft2) / v^(3/4) // Cancel // Apart
```

$$\frac{3\,z\,\mathrm{Dt}[v,x]^2}{16\,v^2}+\mathrm{Dt}[z,\{x,2\}]$$

So the above differential operator das not contain Dt[z,x]. This is valid for any function v, in particular for v = x and v = x^4 :

```
% /. v → x
```

$$\frac{3\,z}{16\,x^2}+\mathrm{Dt}[z,\{x,2\}]$$

```
%% x^2 /. v → x⁴ // Expand
```

$3\,z + x^2\,\mathrm{Dt}[z, \{x, 2\}]$

10.2 Series Expansions

10.2.1 Taylor and Laurent Series

10.2.1.1 Expanding a Function

Series[*expr*, {x,x0,nx}]	expands *expr* around **x = x0** up to n-th power
Series[*expr*, {x,x0,nx}, {y,y0,ny}]	Expansion in **x, y**
Normal[series expression]	removes the remainder term **O[...]**

```
f = Series[Exp[x], {x, 0, 4}]
```

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + O[x]^5$

The remainder term classifies the expression as a series; this is not the same as a polynomial or other combination of functions. In particular, any function added to the series will be replaced automatically by its series:

```
g = f + Exp[2 x]
```

$2 + 3\,x + \frac{5\,x^2}{2} + \frac{3\,x^3}{2} + \frac{17\,x^4}{24} + O[x]^5$

The command **Normal[]** removes the remainder term; the resulting expression consists of common functions.

```
Normal[f] + Exp[2 x]
```

$1 + e^{2\,x} + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24}$

```
Series[Sin[x], {x, 0, 4}]
```

$x - \frac{x^3}{6} + O[x]^5$

```
h = Series[Exp[x], {x, 1, 2}]
```

$e + e\,(x-1) + \frac{1}{2}\,e\,(x-1)^2 + O[x-1]^3$

```
Series[Log[1 + x], {x, 0, 5}]
```

$x - \frac{x^2}{2} + \frac{x^3}{3} - \frac{x^4}{4} + \frac{x^5}{5} + O[x]^6$

```
t = g^2
```

$4 + 12\,x + 19\,x^2 + 21\,x^3 + \frac{217\,x^4}{12} + O[x]^5$

Below we get a Laurent series:

```
Series[Exp[x] / x^2, {x, 0, 4}]
```

$\frac{1}{x^2} + \frac{1}{x} + \frac{1}{2} + \frac{x}{6} + \frac{x^2}{24} + \frac{x^3}{120} + \frac{x^4}{720} + O[x]^5$

```
Series[Sin[x] / x^2, {x, 0, 3}]
```

$$\frac{1}{x}-\frac{x}{6}+\frac{x^3}{120}+O[x]^4$$

```
Series[Exp[Sqrt[x]], {x, 0, 2}]
```

$$1+\sqrt{x}+\frac{x}{2}+\frac{x^{3/2}}{6}+\frac{x^2}{24}+O[x]^{5/2}$$

```
Series[Log[x], {x, 0, 3}]
```

$$\text{Log}[x]+O[x]^4$$

```
Series[Exp[2 x] Log[x], {x, 0, 2}]
```

$$\text{Log}[x]+2\,\text{Log}[x]\,x+2\,\text{Log}[x]\,x^2+O[x]^3$$

```
Series[Exp[1 / x], {x, 0, 2}]
```

$$e^{\frac{1}{x}}$$

$e^{\frac{1}{x}}$ has an essential singularity at $x = 0$. So it is not possible to provide an expansion around this point. But the function $e^{\frac{1}{x}}$ may be expanded arount $x = \infty$:

```
Series[Exp[1 / x], {x, Infinity, 3}]
```

$$1+\frac{1}{x}+\frac{1}{2x^2}+\frac{1}{6x^3}+O\left[\frac{1}{x}\right]^4$$

```
Series[(a + x)^n, {x, 0, 2}]
```

$$a^n+a^{-1+n}\,n\,x+\frac{1}{2}a^{-2+n}\,(-1+n)\,n\,x^2+O[x]^3$$

```
Clear[f, g];
Series[ f[x], {x, 0, 3}]
```

$$f[0]+f'[0]\,x+\frac{1}{2}f''[0]\,x^2+\frac{1}{6}f^{(3)}[0]\,x^3+O[x]^4$$

Functions of several variables may be expanded w.r.t. all variables. The resulting expression may appear as different depending on the sequence of the expansions.

```
Clear[x, y, g]
Series[ g[x, y], {x, 0, 3}, {y, 1, 2} ]
```

$$\left(g[0,1]+g^{(0,1)}[0,1]\,(y-1)+\frac{1}{2}g^{(0,2)}[0,1]\,(y-1)^2+O[y-1]^3\right)+$$
$$\left(g^{(1,0)}[0,1]+g^{(1,1)}[0,1]\,(y-1)+\frac{1}{2}g^{(1,2)}[0,1]\,(y-1)^2+O[y-1]^3\right)x+$$
$$\left(\frac{1}{2}g^{(2,0)}[0,1]+\frac{1}{2}g^{(2,1)}[0,1]\,(y-1)+\frac{1}{4}g^{(2,2)}[0,1]\,(y-1)^2+O[y-1]^3\right)x^2+$$
$$\left(\frac{1}{6}g^{(3,0)}[0,1]+\frac{1}{6}g^{(3,1)}[0,1]\,(y-1)+\frac{1}{12}g^{(3,2)}[0,1]\,(y-1)^2+O[y-1]^3\right)x^3+O[x]^4$$

```
f = Series[ Exp[a x + b y], {x, 0, 3}, {y, 0, 3} ]
```

$$\left(1+b\,y+\frac{b^2y^2}{2}+\frac{b^3y^3}{6}+O[y]^4\right)+\left(a+a\,b\,y+\frac{1}{2}a\,b^2y^2+\frac{1}{6}a\,b^3y^3+O[y]^4\right)x+$$
$$\left(\frac{a^2}{2}+\frac{1}{2}a^2\,b\,y+\frac{1}{4}a^2\,b^2y^2+\frac{1}{12}a^2\,b^3y^3+O[y]^4\right)x^2+$$
$$\left(\frac{a^3}{6}+\frac{1}{6}a^3\,b\,y+\frac{1}{12}a^3\,b^2y^2+\frac{1}{36}a^3\,b^3y^3+O[y]^4\right)x^3+O[x]^4$$

g = Series[Exp[a x + b y], {y, 0, 3}, {x, 0, 3}]

$$\left(1+a\,x+\frac{a^2\,x^2}{2}+\frac{a^3\,x^3}{6}+O[x]^4\right)+\left(b+a\,b\,x+\frac{1}{2}\,a^2\,b\,x^2+\frac{1}{6}\,a^3\,b\,x^3+O[x]^4\right)y+\left(\frac{b^2}{2}+\frac{1}{2}\,a\,b^2\,x+\frac{1}{4}\,a^2\,b^2\,x^2+\frac{1}{12}\,a^3\,b^2\,x^3+O[x]^4\right)y^2+\left(\frac{b^3}{6}+\frac{1}{6}\,a\,b^3\,x+\frac{1}{12}\,a^2\,b^3\,x^2+\frac{1}{36}\,a^3\,b^3\,x^3+O[x]^4\right)y^3+O[y]^4$$

f - g // Simplify

$$\left(\left(-a\,x-\frac{a^2\,x^2}{2}-\frac{a^3\,x^3}{6}+O[x]^4\right)+\left(-a\,b\,x-\frac{1}{2}\left(a^2\,b\right)x^2-\frac{1}{6}\left(a^3\,b\right)x^3+O[x]^4\right)y+\left(-\frac{1}{2}\left(a\,b^2\right)x-\frac{1}{4}\left(a^2\,b^2\right)x^2-\frac{1}{12}\left(a^3\,b^2\right)x^3+O[x]^4\right)y^2+\left(-\frac{1}{6}\left(a\,b^3\right)x-\frac{1}{12}\left(a^2\,b^3\right)x^2-\frac{1}{36}\left(a^3\,b^3\right)x^3+O[x]^4\right)y^3+O[y]^4\right)+\left(a+a\,b\,y+\frac{1}{2}\,a\,b^2\,y^2+\frac{1}{6}\,a\,b^3\,y^3+O[y]^4\right)x+\left(\frac{a^2}{2}+\frac{1}{2}\,a^2\,b\,y+\frac{1}{4}\,a^2\,b^2\,y^2+\frac{1}{12}\,a^2\,b^3\,y^3+O[y]^4\right)x^2+\left(\frac{a^3}{6}+\frac{1}{6}\,a^3\,b\,y+\frac{1}{12}\,a^3\,b^2\,y^2+\frac{1}{36}\,a^3\,b^3\,y^3+O[y]^4\right)x^3+O[x]^4$$

f = Series[Exp[x], {x, 0, 4}]

$$1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+O[x]^5$$

g = f^2

$$1+2\,x+2\,x^2+\frac{4\,x^3}{3}+\frac{2\,x^4}{3}+O[x]^5$$

h = Log[g]

$$2\,x+O[x]^5$$

g = 1/(1 - f)

$$-\frac{1}{x}+\frac{1}{2}-\frac{x}{12}+O[x]^3$$

f = Series[Cos[x], {x, 0, 5}]

$$1-\frac{x^2}{2}+\frac{x^4}{24}+O[x]^6$$

g = D[f, x]

$$-x+\frac{x^3}{6}+O[x]^5$$

h = Integrate[g, x]

$$-\frac{x^2}{2}+\frac{x^4}{24}+O[x]^6$$

f - h

$$1+O[x]^6$$

```
k = 1/f
```

$$1+\frac{x^2}{2}+\frac{5\,x^4}{24}+O[x]^6$$

```
f = Series[Sin[x], {x, 0, 5}]
```

$$x-\frac{x^3}{6}+\frac{x^5}{120}+O[x]^6$$

```
g = f + Sin[x]
```

$$2\,x-\frac{x^3}{3}+\frac{x^5}{60}+O[x]^6$$

```
g = Normal[f]
```

$$x-\frac{x^3}{6}+\frac{x^5}{120}$$

```
g + Sin[x]
```

$$x-\frac{x^3}{6}+\frac{x^5}{120}+\mathrm{Sin}[x]$$

```
Plot[{f, Sin[x]}, {x, 0, π}, Ticks → {{0, π/2, π}, Automatic}, ImageSize → 200]
```



SeriesData::"ssdn" : Attempt to evaluate a series at the number 0.0000641782`. Returning indeterminate.>>

...

The source of the above troubles is the remainder term contained in **f**. **g** no longer contains this, so everything works smoothly:

```
ps = Plot[{g, Sin[x]}, {x, 0, π}, Ticks → {{0, π/2, π}, Automatic}];
pa = Plot[g - Sin[x], {x, 0, π}, Ticks → {{0, π/2, π}, Automatic}];
Show[GraphicsRow[{ps, pa}], ImageSize → 350]
```




```
Solve[ f == 0, x]
```

{}

There is no solution since f is no polynomial but an expression with a remainder term.
g no longer contains it, we get the roots of the resulting fifth orer polynomial:

```
Solve[ g == 0, x]
```

$$\left\{\{x \to 0\}, \left\{x \to -\sqrt{2\left(5 - i\sqrt{5}\right)}\right\}, \left\{x \to \sqrt{2\left(5 - i\sqrt{5}\right)}\right\}, \left\{x \to -\sqrt{2\left(5 + i\sqrt{5}\right)}\right\}, \left\{x \to \sqrt{2\left(5 + i\sqrt{5}\right)}\right\}\right\}$$

10.2.1.2 Substituting Series into Series. Inverse Series

`Series1[x] /. x -> Series2[x]`	Inserts Series_2 into Series_1.
`InverseSeries[Series]`	

```
Series[Sin[x], {x, 0, 5}]
```

$$x - \frac{x^3}{6} + \frac{x^5}{120} + O[x]^6$$

```
% /. x -> Series[Sin[x], {x, 0, 5}]
```

$$x - \frac{x^3}{3} + \frac{x^5}{10} + O[x]^6$$

```
Series[Sin[Sin[x]], {x, 0, 5}]
```

$$x - \frac{x^3}{3} + \frac{x^5}{10} + O[x]^6$$

```
f = Series[Sin[y], {y, 0, 5}]
```

$$y - \frac{y^3}{6} + \frac{y^5}{120} + O[y]^6$$

```
g = InverseSeries[f]
```

$$y + \frac{y^3}{6} + \frac{3\,y^5}{40} + O[y]^6$$

```
k = Series[ArcSin[x], {x, 0, 5}]
```

$$x + \frac{x^3}{6} + \frac{3\,x^5}{40} + O[x]^6$$

The reciprocal of a series

```
1/f
```

$$\frac{1}{y} + \frac{y}{6} + \frac{7\,y^3}{360} + O[y]^4$$

```
g /. y -> f
```

$$y + O[y]^6$$

differs from the series expansion of the inverse function.

```
f = Series[Cot[x], {x, 0, 5}]
```

$$\frac{1}{x} - \frac{x}{3} - \frac{x^3}{45} - \frac{2\,x^5}{945} + O[x]^6$$

```
g = InverseSeries[f]
```

$$\frac{1}{x} - \frac{1}{3\,x^3} + \frac{1}{5\,x^5} - \frac{1}{7\,x^7} + O\left[\frac{1}{x}\right]^8$$

```
h = f /. x -> g
```

$x + O\left[\frac{1}{x}\right]^6$

```
f = Series[1 - Cos[x], {x, 0, 8}]
```

$\frac{x^2}{2} - \frac{x^4}{24} + \frac{x^6}{720} - \frac{x^8}{40320} + O[x]^9$

```
g = InverseSeries[f]
```

$\sqrt{2}\,\sqrt{x} + \frac{x^{3/2}}{6\sqrt{2}} + \frac{3\,x^{5/2}}{80\sqrt{2}} + \frac{5\,x^{7/2}}{448\sqrt{2}} + O[x]^4$

```
f /. x -> g
```

$$x + \left(\frac{1}{18} + \frac{1}{24}\left(-\frac{1}{3} - \sqrt{2}\left(\frac{1}{3\sqrt{2}} + \frac{\sqrt{2}}{3}\right)\right)\right)x^3 + \left(\frac{1}{72} + \frac{1}{24}\left(-\frac{3}{40} - \sqrt{2}\left(\frac{47}{360\sqrt{2}} + \frac{4\sqrt{2}}{45}\right) - \frac{\frac{1}{3\sqrt{2}} + \frac{\sqrt{2}}{3}}{6\sqrt{2}}\right) + \frac{1}{720}\left(\frac{2}{3} + \sqrt{2}\left(\frac{\sqrt{2}}{3} + \sqrt{2}\left(\frac{1}{3} + \sqrt{2}\left(\frac{1}{3\sqrt{2}} + \frac{\sqrt{2}}{3}\right)\right)\right)\right)\right)x^4 + O[x]^{9/2}$$

```
ExpandAll[%]
```

$x + O[x]^{9/2}$

```
f = Series[ Cos[x], {x, 0, 5} ]
```

$1 - \frac{x^2}{2} + \frac{x^4}{24} + O[x]^6$

```
g = InverseSeries[f]
```

$i\,\sqrt{2}\,\sqrt{x-1} - \frac{i\,(x-1)^{3/2}}{6\sqrt{2}} + O[x-1]^{5/2}$

```
f = Series[ Cos[x], {x, Pi/2, 5} ]
```

$-\left(x - \frac{\pi}{2}\right) + \frac{1}{6}\left(x - \frac{\pi}{2}\right)^3 - \frac{1}{120}\left(x - \frac{\pi}{2}\right)^5 + O\left[x - \frac{\pi}{2}\right]^6$

```
f = Series[BesselJ[3, x], {x, 0, 5}]
```

$\frac{x^3}{48} - \frac{x^5}{768} + O[x]^6$

```
f = Series[BesselJ[n, x], {x, 0, 5} ]
```

$x^n\left(\frac{2^{-n}}{\text{Gamma}[1+n]} - \frac{2^{-2-n}\,x^2}{(1+n)\,\text{Gamma}[1+n]} + \frac{2^{-5-n}\,x^4}{(1+n)\,(2+n)\,\text{Gamma}[1+n]} + O[x]^6\right)$

```
f = Series[BesselY[n, x], {x, 0, 5}]
```

$$x^{-n}\left(x^{2n}\left(-\frac{2^{-n}\,\text{Cos}[n\,\pi]\,\text{Gamma}[-n]}{\pi}+\frac{2^{-2-n}\,\text{Cos}[n\,\pi]\,\text{Gamma}[-n]\,x^2}{(1+n)\,\pi}-\frac{2^{-5-n}\,\text{Cos}[n\,\pi]\,\text{Gamma}[-n]\,x^4}{(1+n)\,(2+n)\,\pi}+O[x]^6\right)+\left(-\frac{2^{n}\,\text{Gamma}[n]}{\pi}-\frac{2^{-2+n}\,\text{Gamma}[n]\,x^2}{(-1+n)\,\pi}-\frac{2^{-5+n}\,\text{Gamma}[n]\,x^4}{(-2+n)\,(-1+n)\,\pi}+O[x]^6\right)\right)$$

```
g = f /. n -> 3
```

$$-\frac{16}{\pi\,x^3}-\frac{2}{\pi\,x}-\frac{x}{4\,\pi}+O[x]^3$$

However, he Bessel function of the second kind, $Y_n(x)$, is also well defined for integer order n, e.g. for n = 3:

```
Series[BesselY[3, x], {x, 0, 8}]
```

$$-\frac{16}{\pi\,x^3}-\frac{2}{\pi\,x}-\frac{x}{4\,\pi}+\left(-\frac{11}{288\,\pi}+\frac{\text{EulerGamma}}{24\,\pi}-\frac{\text{Log}[2]}{24\,\pi}+\frac{\text{Log}[x]}{24\,\pi}\right)x^3+\left(\frac{37}{9216\,\pi}-\frac{\text{EulerGamma}}{384\,\pi}+\frac{\text{Log}[2]}{384\,\pi}-\frac{\text{Log}[x]}{384\,\pi}\right)x^5+\left(-\frac{227}{1\,843\,200\,\pi}+\frac{\text{EulerGamma}}{15\,360\,\pi}-\frac{\text{Log}[2]}{15\,360\,\pi}+\frac{\text{Log}[x]}{15\,360\,\pi}\right)x^7+O[x]^9$$

10.2.1.3 Residues

Assume z0 as a n-th order pole of a complex function f(z). In the Laurent series around this singular point

$$f(z)\;=\;\frac{a_{-n}}{(z-z0)^n}+\frac{a_{-(n-1)}}{(z-z0)^{(n-1)}}+\ldots+\frac{a_{-1}}{(z-z0)^n}+\;P(z\text{ - }z0)$$

a_{-1} is called the residue. It is very important in Cauchy's residue theorem.

Residue[*expr*, {x,x0}]	Residue of *expr* at **x = x0**

```
Residue[1 / x, {x, 0}]
```

1

```
Residue[1/x^2, {x, 0}]
```

0

```
Residue[Exp[I 2 x]/x^4, {x, 0}]
```

$$-\frac{4\,i}{3}$$

```
Residue[ Cos[x]/x^3 - Sin[x]^2/x^3, {x, 0}]
```

$$-\frac{3}{2}$$

```
Residue[Gamma[x], {x, -2}]
```

$$\frac{1}{2}$$

```
Limit[ (x + 2) Gamma[x], x -> -2]
```

$$\frac{1}{2}$$

```
Limit[ (x + 3) Gamma[x], x -> -3 ]
```

$-\frac{1}{6}$

```
Residue[BesselJ[2, x] / x^3, {x, 0}]
```

$\frac{1}{8}$

```
LegendreP[2, x]
```

$\frac{1}{2}\left(-1+3\,x^2\right)$

```
Residue[LegendreP[2, x] / x^3, {x, 0} ]
```

$\frac{3}{2}$

```
Residue[ LegendreP[n, x] / x^(n+1), {x, 0} , Assumptions → Element[n, Integers]]
```

$\text{Residue}\left[x^{-1-n}\,\text{LegendreP}[n,x],\{x,0\},\text{Assumptions}\to n\in\text{Integers}\right]$

```
LegendreQ[2, x]
```

$-\frac{3\,x}{2}+\frac{1}{2}\left(-1+3\,x^2\right)\left(-\frac{1}{2}\,\text{Log}[1-x]+\frac{1}{2}\,\text{Log}[1+x]\right)$

```
f = Series[ LegendreQ[2, x], {x, Infinity, 5} ]
```

$-\frac{3}{4}\,i\,\pi\,x^2+\frac{i\,\pi}{4}+\frac{2}{15\,x^3}+\frac{4}{35\,x^5}+O\left[\frac{1}{x}\right]^6$

```
Residue[ LegendreQ[2, x], {x, Infinity} ]
```

0

```
Residue[ LegendreQ[2, x] x^2, {x, Infinity} ]
```

$-\frac{2}{15}$

```
Residue[ BesselY[1, x], {x, 0}]
```

$-\frac{2}{\pi}$

```
Limit[ BesselY[1, x] x, x -> 0]
```

$-\frac{2}{\pi}$

10.3 Limits

10.3.1 Common Limits

```
Limit[expr,  var  -> value]
```

```
Limit[ Sin[2 x] / x,  x -> 0]
```
2

```
Limit[ (Sin[2 x] / x) ^2,  x -> 0]
```
4

```
Limit[ Sin[2 x] ^2 / x,  x -> 0]
```
0

```
Limit[ Sin[2 x] / x^2,  x -> 0]
```
∞

```
Limit[ 1 / Sin[x] - Coth[x],  x -> 0]
```
0

```
Limit[ x / Sin[Pi - x], x -> 0 ]
```
1

```
Limit[ 1 / x,  x -> Infinity]
```
0

```
Clear[g];
g[x_] = Sqrt[x^2 - 4 x] - x;
Limit[ g[x], x -> Infinity ]
```
-2

```
f = (q^a - q^-a) / (q - q^-1)
```
$$\frac{-q^{-a}+q^{a}}{-\frac{1}{q}+q}$$

```
Limit[f, q -> 1]
```
a

```
theta[x_] = Which[x > 0, 1, x == 0, 1 / 2, x < 0, 0 ]
```
$\text{Which}\left[x > 0,\ 1,\ x == 0,\ \frac{1}{2},\ x < 0,\ 0\right]$

```
Limit[theta[x], x -> 0]
```
1

```
Sign[0.1]
```
1

```
Sign[0]
```
0

```
Sign[-1]
```

-1

```
Limit[Sign[x], x -> 0]
```

1

```
Limit[Sign[x], x -> 0.1]
```

1.

10.3.2 Directional Limits

`Limit[expr, x -> x0, Direction ->  1]`	Limit from below
`Limit[expr, x -> x0, Direction -> -1]`	Limit from above

```
Clear[f];
f[x_] = 1 / x
```

$\frac{1}{x}$

```
Limit[f[x], x -> 0, Direction -> -1 ]
```

∞

```
Limit[f[x], x -> 0, Direction ->  1 ]
```

$-\infty$

```
Limit[Sign[x], x -> 0, Direction -> -1 ]
```

1

```
Limit[Sign[x], x -> 0, Direction ->  1 ]
```

-1

```
theta[x_] = Which[x > 0, 1, x == 0, 1 / 2, x < 0, 0 ]
```

$\text{Which}\left[x > 0, 1, x == 0, \frac{1}{2}, x < 0, 0\right]$

```
Limit[theta[x], x -> 0, Direction -> 1]
```

0

```
Limit[ Exp[Tan[x] / Log[Cos[x]]], x -> π / 2, Direction -> -1]
```

ComplexInfinity

```
Limit[ Exp[Tan[x] / Log[Cos[x]]], x -> π / 2, Direction -> 1]
```

0

```
Clear[a, r, it]
it = Integrate[r^2 Exp[-a r^2 ], {r, 0, r}]
```

$$-\frac{e^{-a r^2} r}{2 a} + \frac{\sqrt{\pi}\, \text{Erf}\left[\sqrt{a}\, r\right]}{4 a^{3/2}}$$

```
Limit[it, r -> Infinity]
```

$$\text{Limit}\left[-\frac{e^{-a r^2} r}{2 a} + \frac{\sqrt{\pi}\, \text{Erf}\left[\sqrt{a}\, r\right]}{4 a^{3/2}}, r \to \infty\right]$$

The limit is ruther simple: The first term becomes zero.

Erf[∞] = 1; so the second term gives $\sqrt{\pi}/(4\ a^{3/2})$.

```
a = 3;
```

```
it[[1]] /. r → 5 // N
```

-2.2322×10^{-33}

```
Erf[√a Infinity]
```

1

```
it[[2]] /. r → 5 // N
```

0.0852772

```
0.08527722566220737`
```

0.0852772

```
Sqrt[π] / 4 / a^(3/2) // N
```

0.0852772

10.4 Integration

10.4.1 Analytic Integration

In same cases the result of an indefinite or definite analytic integration or an error message may be wrong. So checks or critical appraisals are strongly recommended. A kind of check may be numerical integration of the same integral.
If the results of these two types of integration differ it is not always clear which result is wrong !

10.4.1.0 Short commands for inputing integrals

Use Esc int Esc to enter $\int$ and Esc dd Esc to enter ⅆ:

$$\int x \, dx$$

$$\frac{x^2}{2}$$

Use the menue "**Palettes**", in the appearing menue "**Other**". Click the **integral sign**.
Color this by pressing simultaneously **Upper case** and the **left shift arrow**.
Then click at the symbol ■□□ in the palette. Insert lower and upper limits of the integral into the empty boxes.

$$\int_a^b x \, dx$$

$$-\frac{9}{2} + \frac{b^2}{2}$$

10.4.1.1 Indefinite Integration

`Integrate[f, x]`	$\int dx\ f$	
`Integrate[f, x, y ]`	$\int dx\ \int dy\ f$	=
`Integrate[Integrate[f,y],x]`	$\int dx\ \int dy\ f$	

```
g = Integrate[Sqrt[x], x]
```

$$\frac{2\, x^{3/2}}{3}$$

```
uous function
```

function uous

```
D[g, x]
```

$$\sqrt{x}$$

```
f = ( (x - 1)^2 (x^2 + 1)^2 )^-1
```

$$\frac{1}{(-1+x)^2 \left(1+x^2\right)^2}$$

```
g = Integrate[f, x]
```

$$\frac{1}{4}\left(\frac{1}{1-x} - \frac{1}{1+x^2} + \text{ArcTan}[x] - 2\,\text{Log}[-1+x] + \text{Log}\left[1+x^2\right]\right)$$

```
h = D[g, x]
```

$$\frac{1}{4}\left(\frac{1}{(1-x)^2}-\frac{2}{-1+x}+\frac{2x}{\left(1+x^2\right)^2}+\frac{1}{1+x^2}+\frac{2x}{1+x^2}\right)$$

```
Together[h]
```

$$\frac{1}{(-1+x)^2\left(1+x^2\right)^2}$$

```
Table[h[[k]], {k, Length[h]}]
```

$$\left\{\frac{1}{4},\ \frac{1}{(1-x)^2}-\frac{2}{-1+x}+\frac{2x}{\left(1+x^2\right)^2}+\frac{1}{1+x^2}+\frac{2x}{1+x^2}\right\}$$

```
Clear[x, f]
```

```
f = (x^3 - 7)^-1
```

$$\frac{1}{-7+x^3}$$

```
Integrate[f, x]
```

$$-\frac{1}{6\times 7^{2/3}}\left(2\sqrt{3}\,\mathrm{ArcTan}\left[\frac{7+2\times 7^{2/3}\,x}{7\sqrt{3}}\right]-2\,\mathrm{Log}\left[7-7^{2/3}\,x\right]+\mathrm{Log}\left[7+7^{2/3}\,x+7^{1/3}\,x^2\right]\right)$$

```
f = (x^3 + x^2 - 7)^-1
```

$$\frac{1}{-7+x^2+x^3}$$

```
Integrate[f, x]
```

$$\mathrm{RootSum}\left[-7+\#1^2+\#1^3\ \&,\ \frac{\mathrm{Log}[x-\#1]}{2\,\#1+3\,\#1^2}\ \&\right]$$

```
ni = N[%]
```

$$0.0889391\,\mathrm{Log}[-1.6311+x]-(0.0444695-0.081883\,i)\,\mathrm{Log}[(1.31555-1.60029\,i)+x]-(0.0444695+0.081883\,i)\,\mathrm{Log}[(1.31555+1.60029\,i)+x]$$

```
ComplexExpand[ni]
```

$$-0.081883\,\mathrm{Arg}[(1.31555-1.60029\,i)+x]+i\,(0.+0.0889391\,\mathrm{Arg}[-1.6311+x]-0.0444695\,\mathrm{Arg}[(1.31555-1.60029\,i)+x]-0.0444695\,\mathrm{Arg}[(1.31555+1.60029\,i)+x])+0.081883\,\mathrm{Arg}[(1.31555+1.60029\,i)+x]+0.0444695\,\mathrm{Log}[(-1.6311+x)^2]-0.0444695\,\mathrm{Log}[2.56091+(1.31555+x)^2]$$

```
ni[[2]]
```

$$(-0.0444695+0.081883\,i)\,\mathrm{Log}[(1.31555-1.60029\,i)+x]$$

```
so = Solve[1/f == 0, x]
```

$$\left\{\left\{x\to\frac{1}{3}\left(-1+\left(\frac{187}{2}-\frac{3\sqrt{3885}}{2}\right)^{1/3}+\left(\frac{1}{2}\left(187+3\sqrt{3885}\right)\right)^{1/3}\right)\right\},\right.$$

$$\left\{x\to-\frac{1}{3}-\frac{1}{6}\left(1+i\sqrt{3}\right)\left(\frac{187}{2}-\frac{3\sqrt{3885}}{2}\right)^{1/3}-\frac{1}{6}\left(1-i\sqrt{3}\right)\left(\frac{1}{2}\left(187+3\sqrt{3885}\right)\right)^{1/3}\right\},$$

$$\left.\left\{x\to-\frac{1}{3}-\frac{1}{6}\left(1-i\sqrt{3}\right)\left(\frac{187}{2}-\frac{3\sqrt{3885}}{2}\right)^{1/3}-\frac{1}{6}\left(1+i\sqrt{3}\right)\left(\frac{1}{2}\left(187+3\sqrt{3885}\right)\right)^{1/3}\right\}\right\}$$

```
g = Integrate[x^3 Log[x], x]
```

$$-\frac{x^4}{16}+\frac{1}{4}x^4\,\mathrm{Log}[x]$$

```
D[g, x]
```

$$x^3\,\mathrm{Log}[x]$$

```
g = Integrate[x^2 / Sqrt[x^2 - 9], x]
```

$$\frac{1}{2}x\sqrt{-9+x^2}+\frac{9}{2}\mathrm{Log}\left[x+\sqrt{-9+x^2}\right]$$

```
D[g, x]
```

$$\frac{x^2}{2\sqrt{-9+x^2}}+\frac{1}{2}\sqrt{-9+x^2}+\frac{9\left(1+\frac{x}{\sqrt{-9+x^2}}\right)}{2\left(x+\sqrt{-9+x^2}\right)}$$

```
Together[%]
```

$$\frac{x^2}{\sqrt{-9+x^2}}$$

```
Clear[a, b, f, g, x]
f = Sqrt[(a^2 - x^2) (b^2 - x^2)] / x
```

$$\frac{\sqrt{\left(a^2-x^2\right)\left(b^2-x^2\right)}}{x}$$

The radicand of the above function is a fourth order polynomial in x. In general, integrals over square roots containing third or fourth order polnomials cannot be expressed as combinations of elmentary functions. One is forced to introduce new transcendental functions, called **elliptic integrals** and **elliptic functions.**
But the integral over the function **f** above can be expressed by elementary functions, so it is called a pseudoelliptic integral:

```
g = Integrate[f, x]
```

$$\left(\left(a^2-x^2\right)\left(b^2-x^2\right)+2\,a\,b\sqrt{-a^2+x^2}\sqrt{-b^2+x^2}\,\mathrm{ArcTanh}\left[\frac{b\sqrt{-a^2+x^2}}{a\sqrt{-b^2+x^2}}\right]-\left(a^2+b^2\right)\sqrt{-a^2+x^2}\sqrt{-b^2+x^2}\,\mathrm{Log}\left[\sqrt{-a^2+x^2}+\sqrt{-b^2+x^2}\right]\right)\Big/\left(2\sqrt{\left(-a^2+x^2\right)\left(-b^2+x^2\right)}\right)$$

```
FullSimplify[g]
```

$$\left((a-x)(b-x)(a+x)(b+x)+\sqrt{-a^2+x^2}\sqrt{-b^2+x^2}\left(2\,a\,b\,\mathrm{ArcTanh}\left[\frac{b\sqrt{-a^2+x^2}}{a\sqrt{-b^2+x^2}}\right]-\left(a^2+b^2\right)\mathrm{Log}\left[\sqrt{-a^2+x^2}+\sqrt{-b^2+x^2}\right]\right)\right)\Big/\left(2\sqrt{(a-x)(b-x)(a+x)(b+x)}\right)$$

```
D[g, x];
```

```
FullSimplify[%]
```

$$\frac{\sqrt{(a-x)(b-x)(a+x)(b+x)}}{x}$$

So the integral is checked by differention. In the literature one finds a simpler expression for this integral:

```
go = 1/2 Sqrt[a^2 - x^2] Sqrt[b^2 - x^2] +
  1/4 (a^2 + b^2) Log[(Sqrt[a^2 - x^2] + Sqrt[b^2 - x^2])/(-Sqrt[a^2 - x^2] + Sqrt[b^2 - x^2])] - 1/2 a b Log[(b Sqrt[a^2 - x^2] + a Sqrt[b^2 - x^2])/(-b Sqrt[a^2 - x^2] + a Sqrt[b^2 - x^2])];
```

```
dg = D[go,x]
```

$$-\frac{x\sqrt{a^2-x^2}}{2\sqrt{b^2-x^2}}-\frac{x\sqrt{b^2-x^2}}{2\sqrt{a^2-x^2}}+\left((a^2+b^2)\left(-\sqrt{a^2-x^2}+\sqrt{b^2-x^2}\right)\left(\frac{-\frac{x}{\sqrt{a^2-x^2}}-\frac{x}{\sqrt{b^2-x^2}}}{-\sqrt{a^2-x^2}+\sqrt{b^2-x^2}}-\frac{\left(\frac{x}{\sqrt{a^2-x^2}}-\frac{x}{\sqrt{b^2-x^2}}\right)\left(\sqrt{a^2-x^2}+\sqrt{b^2-x^2}\right)}{\left(-\sqrt{a^2-x^2}+\sqrt{b^2-x^2}\right)^2}\right)\right)\Big/\left(4\left(\sqrt{a^2-x^2}+\sqrt{b^2-x^2}\right)\right)-\left(a\,b\left(-b\sqrt{a^2-x^2}+a\sqrt{b^2-x^2}\right)\left(\frac{-\frac{b\,x}{\sqrt{a^2-x^2}}-\frac{a\,x}{\sqrt{b^2-x^2}}}{-b\sqrt{a^2-x^2}+a\sqrt{b^2-x^2}}-\frac{\left(\frac{b\,x}{\sqrt{a^2-x^2}}-\frac{a\,x}{\sqrt{b^2-x^2}}\right)\left(b\sqrt{a^2-x^2}+a\sqrt{b^2-x^2}\right)}{\left(-b\sqrt{a^2-x^2}+a\sqrt{b^2-x^2}\right)^2}\right)\right)\Big/\left(2\left(b\sqrt{a^2-x^2}+a\sqrt{b^2-x^2}\right)\right)$$

```
FullSimplify[dg] // Timing
```

$$\left\{0.041272,\ \frac{\sqrt{(a-x)\,(a+x)}\ \sqrt{(b-x)\,(b+x)}}{x}\right\}$$

On differentiating integrals over trigonometric functions, the resulting expression may differ from the original integrand. Nevertheless a check is possible:

```
f = Sin[3 x] Cos[x]^2
```

$\mathrm{Cos}[x]^2\,\mathrm{Sin}[3\,x]$

```
g = Integrate[f, x]
```

$$-\frac{\mathrm{Cos}[x]}{4}-\frac{1}{6}\mathrm{Cos}[3\,x]-\frac{1}{20}\mathrm{Cos}[5\,x]$$

```
h = D[g, x]
```

$$\frac{\mathrm{Sin}[x]}{4}+\frac{1}{2}\mathrm{Sin}[3\,x]+\frac{1}{4}\mathrm{Sin}[5\,x]$$

```
k = h - f
```

$$\frac{\mathrm{Sin}[x]}{4}+\frac{1}{2}\mathrm{Sin}[3\,x]-\mathrm{Cos}[x]^2\,\mathrm{Sin}[3\,x]+\frac{1}{4}\mathrm{Sin}[5\,x]$$

```
Simplify[k]
```

0

On integrating functions of several variables over several of these may lead to different expressions depending on the order of the integration. This is not incorrect, since the derivative of the difference is zero.

```
Clear[x, y, z, r]
r = Sqrt[x^2 + y^2 + z^2]
```

$\sqrt{x^2+y^2+z^2}$

```
gxy = Integrate[x^2/r, x, y]
```

$$\frac{1}{18}\left(-2x^3+6xz^2+3xy\sqrt{x^2+y^2+z^2}-6z^3\text{ArcTan}\left[\frac{x}{z}\right]+6z^3\text{ArcTan}\left[\frac{xy}{z\sqrt{x^2+y^2+z^2}}\right]-3y\left(y^2+3z^2\right)\text{Log}\left[x+\sqrt{x^2+y^2+z^2}\right]+6x^3\text{Log}\left[y+\sqrt{x^2+y^2+z^2}\right]\right)$$

```
gyx = Integrate[x^2/r, y, x]
```

$$\frac{1}{18}\left(y^3+6yz^2+3xy\sqrt{x^2+y^2+z^2}-6z^3\text{ArcTan}\left[\frac{y}{z}\right]+6z^3\text{ArcTan}\left[\frac{xy}{z\sqrt{x^2+y^2+z^2}}\right]-3y^3\text{Log}\left[x+\sqrt{x^2+y^2+z^2}\right]-9yz^2\text{Log}\left[x+\sqrt{x^2+y^2+z^2}\right]+6x^3\text{Log}\left[y+\sqrt{x^2+y^2+z^2}\right]\right)$$

```
dg = gxy - gyx
```

$$\frac{1}{18}\left(-y^3-6yz^2-3xy\sqrt{x^2+y^2+z^2}+6z^3\text{ArcTan}\left[\frac{y}{z}\right]-6z^3\text{ArcTan}\left[\frac{xy}{z\sqrt{x^2+y^2+z^2}}\right]+3y^3\text{Log}\left[x+\sqrt{x^2+y^2+z^2}\right]+9yz^2\text{Log}\left[x+\sqrt{x^2+y^2+z^2}\right]-6x^3\text{Log}\left[y+\sqrt{x^2+y^2+z^2}\right]\right)+\frac{1}{18}\left(-2x^3+6xz^2+3xy\sqrt{x^2+y^2+z^2}-6z^3\text{ArcTan}\left[\frac{x}{z}\right]+6z^3\text{ArcTan}\left[\frac{xy}{z\sqrt{x^2+y^2+z^2}}\right]-3y\left(y^2+3z^2\right)\text{Log}\left[x+\sqrt{x^2+y^2+z^2}\right]+6x^3\text{Log}\left[y+\sqrt{x^2+y^2+z^2}\right]\right)$$

```
Simplify[dg]
```

$$\frac{1}{18}\left(-2x^3-y^3+6xz^2-6yz^2-6z^3\text{ArcTan}\left[\frac{x}{z}\right]+6z^3\text{ArcTan}\left[\frac{y}{z}\right]\right)$$

```
D[dg, x, y] // Together
```

0

Integration by hand gives:

```
g = x y r / 6 + z^3 ArcTan[x y / (z r)] / 3 -
      (y^3 + 3 y z^2) Log[x + r] / 6 + x^3 Log[y + r] / 3
```

$$\frac{1}{6}xy\sqrt{x^2+y^2+z^2}+\frac{1}{3}z^3\text{ArcTan}\left[\frac{xy}{z\sqrt{x^2+y^2+z^2}}\right]-\frac{1}{6}\left(y^3+3yz^2\right)\text{Log}\left[x+\sqrt{x^2+y^2+z^2}\right]+\frac{1}{3}x^3\text{Log}\left[y+\sqrt{x^2+y^2+z^2}\right]$$

```
f = D[g, x, y] // Together
```

$$\frac{x^2}{\sqrt{x^2+y^2+z^2}}$$

```
gxy = Integrate[x^2 / r^2, x, y]
```

$$\int \frac{x^2 \text{ArcTan}\left[\frac{y}{\sqrt{x^2+z^2}}\right]}{\sqrt{x^2+z^2}} \, dx$$

In fact, a more detailed and lengthy analysis not given here shows that the above integral cannot be expressed by elementary functions.

```
rr = Sqrt[x^2 + y^2]
```

$$\sqrt{x^2 + y^2}$$

```
k = Integrate[rr^-3, x, y]
```

$$-\frac{\sqrt{x^2 + y^2}}{x\, y}$$

```
D[k, x, y] // Together
```

$$\frac{1}{\left(x^2 + y^2\right)^{3/2}}$$

```
f = Sqrt[1 + x^6] / x
```

$$\frac{\sqrt{1 + x^6}}{x}$$

```
Integrate[f, x]
```

$$\frac{\sqrt{1 + x^6}}{3} - \frac{1}{3} \text{ArcTanh}\left[\sqrt{1 + x^6}\right]$$

```
Clear[a, x];
f = ArcSinh[a / x];
g = Integrate[f, x] // ExpandAll // PowerExpand
```

$$x \, \text{ArcSinh}\left[\frac{a}{x}\right] + \frac{a \sqrt{a^2 + x^2} \, \text{Log}\left[x + \sqrt{a^2 + x^2}\right]}{\sqrt{1 + \frac{a^2}{x^2}} \, x}$$

```
h = D[g, x] // FullSimplify
```

$$\text{ArcSinh}\left[\frac{a}{x}\right]$$

One may perform the integration by hand by partial integration. In this way one gets:

$$\int \text{Arsh(a/x) dx} \quad = x \ \text{Arsh(a/x)} \quad + \quad \int \text{dx} \ \ a/x \ \ 1 / \text{sqrt}[1 + (a/x)\text{^}2]$$

$$= x \ \text{Arsh(a/x)} \quad + \quad a \int \text{dx} / \text{sqrt}(a\text{^}2 + x\text{^}2)$$

$$= x \ \text{Arsh(a/x)} \quad + \quad a \ \ \text{Arsh(x/a)}$$

```
g1 = x ArcSinh[a / x] + a ArcSinh[x / a];
h1 = D[g1, x]
```

$$-\frac{a}{\sqrt{1 + \frac{a^2}{x^2}} \, x} + \frac{1}{\sqrt{1 + \frac{x^2}{a^2}}} + \text{ArcSinh}\left[\frac{a}{x}\right]$$

```
Simplify[h1, (x >= 0 && a >= 0)]
```

$$\text{ArcSinh}\left[\frac{a}{x}\right]$$

Indefinite integrals over discontinuous functions, as e.g. **Abs[x], Sign[x]** are not done.
A workaround is to define them via the **UnitStep[x]** function:

```
Integrate[Abs[x], x]
```

$\int \mathrm{Abs}[x]\, dx$

```
Integrate[2 x UnitStep[x] - x, x]
```

$-\frac{x^2}{2} + x^2\,\mathrm{UnitStep}[x]$

```
Show[GraphicsRow[
  {Plot[Abs[x], {x, -1, 1}, AxesLabel → {"x", "Abs[x]"}], Plot[2 x UnitStep[x] - x,
    {x, -1, 1}, AxesLabel → {"x", "2 x UnitStep[x] - x"}]}], ImageSize → 400]
```




```
NIntegrate[Abs[x], {x, -1, 2}]
```

2.5

```
Integrate[2 x UnitStep[x] - x, {x, -1, 2}]
```

$\frac{5}{2}$

10.4.1.2 Definite Integrals

Limits of integration may be inserted by substitutions after the indefinite integration has been performed.

```
g = Integrate[Sin[x], x]
```

-Cos[x]

```
gt = g /. x -> Pi/2  -  g /. x -> 0
```

Sin[1]

```
gt = (g /. x -> Pi/2) - (g /. x -> 0)
```

1

Integration limits can be given in the command `Integrate[]`

`Integrate[f, {x,x0,x1}]`	$\int_{x0}^{x1} f(x)\, dx$
`Integrate[f, {x,x0,x1}, {y,y0,y1}]`	$\int_{x0}^{x1}\int_{y0}^{y1} f[x, y]\, dy\, dx$

```
Integrate[ Sin[x], {x,0,π/2} ]
```

1

```
Integrate[ Sin[.1 x], {x,0,π/2} ]
```

0.123117

```
f = x^3 y^2 z
```

$x^3\, y^2\, z$

```
Integrate[f, {x,a,b}, {y,c,d}, {z,e,m}]
```

$$-\frac{1}{24}\left(a^4-b^4\right)\left(c^3-d^3\right)\left(e^2-m^2\right)$$

```
Integrate[Sin[x]/x, {x,0,b} ]
```

SinIntegral[b]

```
Integrate[((x - 1) Log[x]^(-1) - x) / Log[x], {x, 0, 1}]
```

$-1+\text{Log}[2]$

```
Integrate[1/Log[x]^2 - x/(1 - x)^2, {x, 0, 1}]
```

$$-\frac{1}{2}+\text{EulerGamma}$$

```
N[%] - NIntegrate[1 / Log[x]^2 - x / (1 - x)^2, {x, 0, 1}]
```

1.00711×10^{-13}

Volume of an octant of a homogeneous tri-axial ellipsoid.
Octant of an ellipsoid:

$$\frac{x^2}{a^2}+\frac{y^2}{b^2}+\frac{z^2}{c^2}=1$$

$$0 \le x \le a,\ 0 \le y \le b,\ 0 \le z \le c.$$

Earlier versions of *Mathematica* could calculate the volume of triaxial ellipsoid rather easily. Present versions need some assistance by the user.

```
vz = Integrate[1,
    {z, 0, c Sqrt[1 - (x / a)^2 - (y / b)^2] } ]
```

$$c\sqrt{1-\frac{x^2}{a^2}-\frac{y^2}{b^2}}$$

```
vy = Integrate[vz, {y, 0, b Sqrt[1 - (x / a)^2]},
  Assumptions -> a > 0 && b > 0 && c > 0 && a > x > 0 ]
```

$$\frac{b\,c\,\pi\,(a-x)\,(a+x)}{4\,a^2}$$

```
v = Integrate[%, {x, 0, a}]
```

$$\frac{1}{6}\,a\,b\,c\,\pi$$

Another workaround is to make the usual trigonometric substitution for the variable y:

```
sy = y → b (1 - x²/a²)^(1 / 2) Sin[ϕ] ;
```

```
vvy = vz /. sy
```

$$c\sqrt{1-\frac{x^2}{a^2}-\left(1-\frac{x^2}{a^2}\right)\text{Sin}[\phi]^2}$$

```
dy = D[y /. sy, ϕ]
```

$$b\sqrt{1-\frac{x^2}{a^2}}\ \text{Cos}[\phi]$$

```
vy = Integrate[vvy dy, {ϕ, 0, π / 2}]
```

$$\frac{b\, c\, \pi\, (a^2 - x^2)}{4\, a^2}$$

```
Integrate[vy, {x, 0, a}]
```

$$\frac{1}{6} a\, b\, c\, \pi$$

10.4.1.3 Infinite Integrals

```
Integrate[ 1/x^2 , {x, 2, Infinity} ]
```

$$\frac{1}{2}$$

```
Integrate[ 1 / x , {x, 2, Infinity} ]
```

Integrate::idiv: Integral of $\frac{1}{x}$ does not converge on $\{2, \infty\}$. ≫

$$\int_2^{\infty} \frac{1}{x}\, dx$$

Integrate::"idiv" : Integral of $\frac{1}{x}$ does not converge on [2,∞]. >>

```
Integrate[Sin[x] / x, {x, 0, Infinity} ]
```

$$\frac{\pi}{2}$$

```
Integrate[ (Sin[x] / x) ^2, {x, 0, Infinity} ]
```

$$\frac{\pi}{2}$$

```
Integrate[Sin[x] ^3 / x , {x, 0, Infinity} ]
```

$$\frac{\pi}{4}$$

```
Integrate[Sin[x]/x^2 , {x,0,Infinity} ]
```

Integrate::idiv: Integral of $\frac{\text{Sin}[x]}{x^2}$ does not converge on $\{0, \infty\}$. ≫

$$\int_0^{\infty} \frac{\text{Sin}[x]}{x^2}\, dx$$

```
Clear[x, b, f]
```

$$f[x\_, b\_] = \frac{x^2\, b\, e^{b x}}{(e^{b x} + 1)^2};$$

```
Integrate[f[x, b], {x, 0, ∞}]
```

$$\text{ConditionalExpression}\left[\frac{\pi^2}{6\, b^2}, \left(\text{Re}\left[(-1)^{\frac{1}{b}}\right] \geq 1 \,||\, \text{Re}\left[(-1)^{\frac{1}{b}}\right] \leq 0 \,||\, (-1)^{\frac{1}{b}} \notin \text{Reals}\right) \,\&\&\, \text{Re}[b] > 0\right]$$

```
Simplify[%, b > 0]
```

$$\text{ConditionalExpression}\left[\frac{\pi^2}{6\, b^2}, \text{Re}\left[(-1)^{\frac{1}{b}}\right] \geq 1 \,||\, \text{Re}\left[(-1)^{\frac{1}{b}}\right] \leq 0 \,||\, (-1)^{\frac{1}{b}} \notin \text{Reals}\right]$$

```
Integrate[f[x, b], {x, 0, ∞}, Assumptions → {b > 0}]
```

$$\frac{\pi^2}{6\,b^2}$$

```
Integrate[f[x, b], {x, 0, ∞}, Assumptions → {Re[b] > 0}]
```

$$\text{ConditionalExpression}\left[\frac{\pi^2}{6\,b^2},\ \text{Im}\left[(-1)^{\frac{1}{b}}\right] \neq 0\ ||\ \text{Re}\left[(-1)^{\frac{1}{b}}\right] \geq 1\ ||\ \text{Re}\left[(-1)^{\frac{1}{b}}\right] \leq 0\right]$$

```
Integrate[1 / 2 + 1 / 2 Erf[z], {z, -Infinity, 0}]
```

$$\frac{1}{2\sqrt{\pi}}$$

Erf[z] is the error function $(2/\sqrt{\pi}\,)\int_0^z e^{-x^2}\,dx$.

10.4.1.3.1 Workarounds, where *Mathematica* does not succed without assistance

$$\text{ic} = \int_0^\infty \frac{\text{Cos}[a\,x]\ \text{Sin}[b\,x]\ \text{Sinh}[(-c+d)\,x]}{x\,\text{Sinh}[d\,x]}\,dx$$

```
$Aborted
```

The algorithms of *Mathematica* do not succeed. So a simpler intgral is done at first.
The product of trigonometric functions is decomposed:

```
Clear[a, b, x, fx]
fx = Cos[a x ] Sin[b x]
```

```
Cos[a x] Sin[b x]
```

```
fxd = TrigReduce[fx] // Expand
```

$$-\frac{1}{2}\text{Sin}[a\,x-b\,x]+\frac{1}{2}\text{Sin}[a\,x+b\,x]$$

```
α1 = fxd[[2, 2, 1]] / x // Cancel
```

```
a + b
```

```
α2 = fxd[[1, 2, 1]] / x // Cancel
```

```
a - b
```

A simpler integrand is used at the start. Later on that of **ic** is obtained by integrating both sides w.r.t. α and identifying some other parameters.

```
fi =  Cos[x α ] Sinh[β x] / Sinh[d x]
```

```
Cos[x α] Csch[d x] Sinh[x β]
```

```
in = Integrate[fi, {x, 0, Infinity} ,
  Assumptions -> d > β > 0 && Element[α, Reals]]
```

$$\frac{\pi\,\text{Sin}\left[\frac{\pi\beta}{d}\right]}{2\,d\,\text{Cos}\left[\frac{\pi\beta}{d}\right]+2\,d\,\text{Cosh}\left[\frac{\pi\alpha}{d}\right]}$$

```
ia = Integrate[fi, {α, 0, α}] /. {β → d - c}
```

$$\frac{\text{Csch}[d\,x]\ \text{Sin}[x\,\alpha]\ \text{Sinh}[(-c+d)\,x]}{x}$$

```
iaf = 1 / 2 (ia /. α → α1) - 1 / 2 (ia /. α → α2) // Together // Simplify
```

$$\frac{\text{Cos}[a\,x]\ \text{Csch}[d\,x]\ \text{Sin}[b\,x]\ \text{Sinh}[(-c+d)\,x]}{x}$$

So this is the integrand of the wanted integral ic. The same operations as above must be applied to **in**:

```
ir = Integrate[in, {α, 0, α}, Assumptions → α > 0 && β > 0 && d > 0]
```

$$\text{ConditionalExpression}\left[\text{ArcTan}\left[\text{Tan}\left[\frac{\pi\beta}{2\,d}\right]\text{Tanh}\left[\frac{\pi\alpha}{2\,d}\right]\right],\ \text{Cos}\left[\frac{\pi\beta}{d}\right]\neq -1\ \&\&\ \text{Cosh}\left[\frac{\pi\alpha}{d}\right]>1\ \&\&\ \text{Cos}\left[\frac{\pi\beta}{2\,d}\right]\neq 0\right]$$

```
irr = ir /. {β → d - c}
```

$$\text{ConditionalExpression}\left[\text{ArcTan}\left[\text{Tan}\left[\frac{(-c+d)\,\pi}{2\,d}\right]\text{Tanh}\left[\frac{\pi\alpha}{2\,d}\right]\right],\ \text{Cos}\left[\frac{(-c+d)\,\pi}{d}\right]\neq -1\ \&\&\ \text{Cosh}\left[\frac{\pi\alpha}{d}\right]>1\ \&\&\ \text{Cos}\left[\frac{(-c+d)\,\pi}{2\,d}\right]\neq 0\right]$$

```
irf = 1/2 (irr /. α → α1) - 1/2 (irr /. α → α2) // Together // Simplify
```

$$\text{ConditionalExpression}\left[\frac{1}{2}\left(-\text{ArcTan}\left[\text{Cot}\left[\frac{c\,\pi}{2\,d}\right]\text{Tanh}\left[\frac{(a-b)\,\pi}{2\,d}\right]\right]+\text{ArcTan}\left[\text{Cot}\left[\frac{c\,\pi}{2\,d}\right]\text{Tanh}\left[\frac{(a+b)\,\pi}{2\,d}\right]\right]\right),\ \text{Sin}\left[\frac{c\,\pi}{2\,d}\right]\neq 0\ \&\&\ \text{Cos}\left[\frac{c\,\pi}{d}\right]\neq 1\ \&\&\ \text{Cosh}\left[\frac{(a-b)\,\pi}{d}\right]>1\ \&\&\ \text{Cosh}\left[\frac{(a+b)\,\pi}{d}\right]>1\right]$$

```
ExpToTrig[irf]
```

$$\text{ConditionalExpression}\left[-\frac{1}{2}\text{ArcTan}\left[\text{Cot}\left[\frac{c\,\pi}{2\,d}\right]\text{Tanh}\left[\frac{(a-b)\,\pi}{2\,d}\right]\right]+\frac{1}{2}\text{ArcTan}\left[\text{Cot}\left[\frac{c\,\pi}{2\,d}\right]\text{Tanh}\left[\frac{(a+b)\,\pi}{2\,d}\right]\right],\ \text{Sin}\left[\frac{c\,\pi}{2\,d}\right]\neq 0\ \&\&\ \text{Cos}\left[\frac{c\,\pi}{d}\right]\neq 1\ \&\&\ \text{Cosh}\left[\frac{(a-b)\,\pi}{d}\right]>1\ \&\&\ \text{Cosh}\left[\frac{(a+b)\,\pi}{d}\right]>1\right]$$

$$\textbf{irs = FullSimplify}\left[\%,\ \text{Sin}\left[\frac{c\,\pi}{d}\right]>0\right]$$

$$\text{ConditionalExpression}\left[\frac{1}{2}\left(-\text{ArcTan}\left[\text{Cot}\left[\frac{c\,\pi}{2\,d}\right]\text{Tanh}\left[\frac{(a-b)\,\pi}{2\,d}\right]\right]+\text{ArcTan}\left[\text{Cot}\left[\frac{c\,\pi}{2\,d}\right]\text{Tanh}\left[\frac{(a+b)\,\pi}{2\,d}\right]\right]\right),\ \text{Sin}\left[\frac{c\,\pi}{2\,d}\right]\neq 0\ \&\&\ \text{Cos}\left[\frac{c\,\pi}{d}\right]\neq 1\ \&\&\ \text{Cosh}\left[\frac{(a-b)\,\pi}{d}\right]>1\ \&\&\ \text{Cosh}\left[\frac{(a+b)\,\pi}{d}\right]>1\right]$$

```
su = {a → .37, b → 1.23, c → .79, d → 3.21};
```

```
Sin[c π / d] /. su
```

0.698404

$$\textbf{NIntegrate}\left[\frac{\text{Cos}[a\,x]\ \text{Sin}[b\,x]\ \text{Sinh}[(-c+d)\,x]}{x\,\text{Sinh}[d\,x]}\ /.\ su,\ \{x,0,\text{Infinity}\}\right]$$

NIntegrate::eerrprec :
Catastrophic loss of precision in the global error estimate due to insufficient WorkingPrecision or divergent integral. ≫

$$\text{NIntegrate}\left[\frac{\text{Cos}[a\,x]\ \text{Sin}[b\,x]\ \text{Sinh}[(-c+d)\,x]}{x\,\text{Sinh}[d\,x]}\ /.\ su,\ \{x,0,\infty\}\right]$$

```
irf /. su
```

0.894002

```
irs /. su
```

0.894002

10.4.1.4 Complex Integration

For integration in the complex plane the path of integration may be given as a sequence of line elements:

```
Integrate[1 / z, {z, -1, I, 1 / 2}]
```

```
- ⅈ π - Log[2]
```

```
N[%]
```

```
- 0.693147 - 3.14159 ⅈ
```

```
NIntegrate[1 / z, {z, -1, I, 1 / 2}] // Chop
```

```
- 0.693147 - 3.14159 ⅈ
```

```
Integrate[1 / z, {z, -1, -I, 1 / 2}]
```

```
ⅈ π - Log[2]
```

```
N[%]
```

```
- 0.693147 + 3.14159 ⅈ
```

```
NIntegrate[1 / z, {z, -1, -I, 1 / 2}]
```

```
- 0.693147 + 3.14159 ⅈ
```

```
Integrate[1 / z, {z, -1, I, 1}]
```

```
- ⅈ π
```

```
N[%]
```

```
0. - 3.14159 ⅈ
```

```
NIntegrate[1 / z, {z, -1, I, 1}] // Chop
```

```
0. - 3.14159 ⅈ
```

Numeric and analytic integration give the same result.

```
NIntegrate[1 / z, {z, -1, I, 2}]
```

```
0.693147 - 3.14159 ⅈ
```

10.4.1.5 Integrals Leading to Multivalued Functions

Integrals leading to multivalued functions as, for example, logarithms or powers with fractional exponents require special care. Version 5 and newer versions are better prepared as earlier versions of Mathemtica to deal with that problem.   So the results below are correct.

```
ia = Integrate[1 / z, {z, -1, -I, 1, I, -1}]
```

```
2 ⅈ π
```

```
N[%] // Chop
```

```
0. + 6.28319 ⅈ
```

```
in = NIntegrate[1 / z, {z, -1, -I, 1, I, -1}] // Chop
```

```
0. + 6.28319 ⅈ
```

10.4.1.6 Options for Integrate[]

```
?? Integrate
```

Integrate[f, x] gives the indefinite integral $\int f\, dx$.

Integrate[f, {x, x_{min}, x_{max}}] gives the definite integral $\int_{x_{min}}^{x_{max}} f\, dx$.

Integrate[f, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}, …] gives the multiple integral $\int_{x_{min}}^{x_{max}} dx \int_{y_{min}}^{y_{max}} dy \ldots f$.

Integrate[f, {x, y, …} ∈ reg] integrates over the geometric region reg. >>

```
Attributes[Integrate] = {Protected, ReadProtected}

Options[Integrate] =
 {Assumptions :→ $Assumptions, GenerateConditions → Automatic, PrincipalValue → False}
```

`Integrate[Sin[a x] / x, {x, 0, Infinity}]`

ConditionalExpression[$\frac{1}{2} \pi$ Sign[a], a ∈ Reals]

`Integrate[Sin[a x] / x, {x, 0, Infinity}, Assumptions -> {a > 0}]`

$$\frac{\pi}{2}$$

`?? GenerateConditions`

GenerateConditions is an option for Integrate, Sum, and similar functions that specifies whether explicit conditions on parameters should be generated in the result >>

```
Attributes[GenerateConditions] = {Protected}
```

`Integrate[Exp[-a x^2], {x, 0, Infinity}]`

ConditionalExpression[$\frac{\sqrt{\pi}}{2 \sqrt{a}}$, Re[a] > 0]

`Integrate[Exp[-a x^2], {x, 0, Infinity}, GenerateConditions -> False]`

$$\frac{\sqrt{\pi}}{2 \sqrt{a}}$$

`Integrate[Sin[a x] / x, {x, 0, Infinity}, GenerateConditions -> False]`

$$\frac{a \pi}{2 \sqrt{a^2}}$$

This is correct for real a, only. But the flow of the calculation is not interrupted by an **`If[]`**.

`?? Assumptions`

Assumptions is an option for functions such as Simplify, Refine, and Integrate that specifies default assumptions to be made about symbolic quantities. >>

```
Attributes[Assumptions] = {Protected}
```

`Integrate[E^(- a x^2), {x, 0, Infinity}]`

ConditionalExpression[$\frac{\sqrt{\pi}}{2 \sqrt{a}}$, Re[a] > 0]

`Integrate[E^(- a x^2), {x, 0, Infinity}, Assumptions → {a > 0}]`

$$\frac{\sqrt{\pi}}{2 \sqrt{a}}$$

```
Integrate[Sin[t*u - x] * Exp[-a*u^2], {u, -Infinity, Infinity}]
```

$$\int_{-\infty}^{\infty}\Big(e^{-a u^2}\,\text{Sin}[4 u] + e^{-a u^2}(-1+12 u)\,\text{Cos}[4 u]\,x + e^{-a u^2}\Big(19 u\,\text{Cos}[4 u] - \frac{1}{2}(-1+12 u)^2\,\text{Sin}[4 u]\Big)x^2 + e^{-a u^2}\Big(21 u\,\text{Cos}[4 u] - \frac{1}{6}(-1+12 u)^3\,\text{Cos}[4 u] - 19 u(-1+12 u)\,\text{Sin}[4 u]\Big)x^3 + e^{-a u^2}\Big(\frac{217}{12}u\,\text{Cos}[4 u] - \frac{19}{2}u(-1+12 u)^2\,\text{Cos}[4 u] + \frac{1}{24}(-1+12 u)^4\,\text{Sin}[4 u] - \frac{1}{2}\big(361 u^2 + 42 u(-1+12 u)\big)\,\text{Sin}[4 u]\Big)x^4 + O[x]^5\Big)\,du$$

```
Integrate[Sin[t*u - x] * Exp[-a*u^2],
 {u, -Infinity, Infinity}, Assumptions → {Element[t, Reals], a > 0}]
```

$$\text{Integrate}\Big[e^{-a u^2}\,\text{Sin}[4 u] + e^{-a u^2}(-1+12 u)\,\text{Cos}[4 u]\,x + e^{-a u^2}\Big(19 u\,\text{Cos}[4 u] - \frac{1}{2}(-1+12 u)^2\,\text{Sin}[4 u]\Big)x^2 + e^{-a u^2}\Big(21 u\,\text{Cos}[4 u] - \frac{1}{6}(-1+12 u)^3\,\text{Cos}[4 u] - 19 u(-1+12 u)\,\text{Sin}[4 u]\Big)x^3 + e^{-a u^2}\Big(\frac{217}{12}u\,\text{Cos}[4 u] - \frac{19}{2}u(-1+12 u)^2\,\text{Cos}[4 u] + \frac{1}{24}(-1+12 u)^4\,\text{Sin}[4 u] - \frac{1}{2}\big(361 u^2 + 42 u(-1+12 u)\big)\,\text{Sin}[4 u]\Big)x^4 + O[x]^5,\ \{u, -\infty, \infty\},\ \text{Assumptions} \to \Big\{4 + 12 x + 19 x^2 + 21 x^3 + \frac{217 x^4}{12} + O[x]^5 \in \text{Reals},\ a > 0\Big\}\Big]$$

`Element[x,dom]`	declares **x** as element of ***dom***

Domains are:

`Algebraics`	algebraic numbers
`Booleans`	`True` or `False`
`Complexes`	complex numbers
`Integers`	integers
`Primes`	prime numbers
`Rationals`	rational numbers
`Reals`	real numbers

10.4.1.7 **Cauchy principal value (=** Cauchyscher Hauptwert)

```
Integrate[Cos[x] / x, {x, -1, 2}]
```

Integrate::idiv: Integral of $\frac{\text{Cos}[x]}{x}$ does not converge on {-1, 2}. ≫

$$\int_{-1}^{2}\frac{\text{Cos}[x]}{x}\,dx$$

The integrand above has a first order pole at x = 0. The corresponding Riemann integral does not exist. For such integrals another definition of the integral may be used called the **Cauchy principal value.** This is defined for integrands, which contain a first order pole so that the Riemann integral no longer exists. The integrands discussed here have such a pole at t = 0. So the Cauchy principal value (PV) is defined as the following limit (a < 0, b > 0):

$$\text{PV}\int_a^b f(t)/t\;dt \;:=$$

$$\lim_{\epsilon\to 0}\Big[\int_a^{-\epsilon} f(t)/t\;dt + \int_\epsilon^b f(t)/t\;dt\Big].$$

If the singular point is at $t = t_0$ then t must be replaced by $t - t_0$ and the integration limits must be changed according to $-\varepsilon \to t_0 - \epsilon$, $\varepsilon \to t_0 + \epsilon$ respectively.

The integral as defined above is not a Rieman integral, so it is denoted as PV (= principal value).

The contributions of the two singular branches cancel for the particular limit defined. This cannot occur, if the particular form of the limit is not observed. For example, a limit

$$\lim_{\epsilon\to 0}\left[\int_a^{-\epsilon} f(t)/t\,dt + \int_{\epsilon^2}^b f(t)/t\,dt\right]$$

would not exist. Whereas this is a prerequisite for a Rieman integral, where an arbitrary decomposition of the interval of integration must be possible, i.e. the widths of the stripes approximating the integral are arbitrary.

An equivalent method to compute the Cauchy principle value uses complex integration. On gets finite values for the integral by choosing an integration path in the complex plane. The principal value is obtained by taking the arithmetic mean of two such complex integrals. The first (second) one uses a path C_U (C_L) passing above (below) the real singular point. Otherwise these paths are quite arbitrary except there must not be any other singularity between the original real path and C_U and C_L. In most cases the paths C_U and C_L may consist of straight lines starting at a, b respectively and meeting at a complex number located roughly above, below the singular point respectively.

$$\text{PV} \int_a^b f(t)/t\,dt := \frac{1}{2}\left[\int_{a,C_U}^b f(t)/t\,dt + \int_{a,C_L}^b f(t)/t\,dt\right]$$

```
?? PrincipalValue
```

PrincipalValue is an option for Integrate that specifies whether the Cauchy principal value should be found for a definite integral >>

```
Attributes[PrincipalValue] = {Protected}
```

Principal Value = Cauchyscher Hauptwert

```
Integrate[1 / x, {x, -1, 2}, PrincipalValue -> True]

Log[2]
```

or by complex integration

```
iu = Integrate[1 / x, {x, -1, I, 2}]
il = Integrate[1 / x, {x, -1, -I, 2}]
(iu + il) / 2

-i π + Log[2]

i π + Log[2]

Log[2]

Integrate[Cos[x] / x, {x, -1, 2}, PrincipalValue -> True]

-CosIntegral[1] + CosIntegral[2]

N[%] // Chop

0.0855769

Clear[t]

Integrate[Sin[t] / t^2, {t, -1, 2}, PrincipalValue -> True]

-CosIntegral[1] + CosIntegral[2] + Sin[1] - Sin[2]/2

N[%] // Chop

0.472399
```

```
Clear[x, x0, x1, xp]
Integrate[1 / (x - xp), {x, x0, x1}, Assumptions → (0 < xp < x0 < x1) ]
```

$\text{Log}\left[\frac{x1 - xp}{x0 - xp}\right]$

```
Integrate[1 / (x - xp), {x, x0, x1}, Assumptions → (0 < x0 < x1 < xp) ]
```

$\text{Log}\left[\frac{x1 - xp}{x0 - xp}\right]$

```
Integrate[1 / (x - xp), {x, x0, x1}, Assumptions → (0 < x0 < xp < x1) ]
```

Integrate::idiv: Integral of $\frac{1}{x - xp}$ does not converge on {x0, x1}. >>

$\text{Integrate}\left[\frac{1}{x - xp}, \{x, x0, x1\}, \text{Assumptions} \to 0 < x0 < xp < x1, \text{PrincipalValue} \to \text{True}\right]$

$\text{Log}\left[\frac{-x1 + xp}{x0 - xp}\right]$

The corresponding numeric integrations are done in § 10.4.2.3 .

10.4.2 Numeric Integration

NIntegrate[f, {x,x_0,x_1}]	$\int_{x0}^{x1} f(x)\, dx$
NIntegrate[f, {x,x_0,x_1}, {y,y_0,y_1}]	$\int_{x0}^{x1}\int_{y0}^{y1} f[x, y]\, dy\, dx$

```
NIntegrate[ Sin[x], {x, 0, π / 2} ]
1.

NIntegrate[ Sin[20 x], {x, 0, π / 2} ]
```

NIntegrate::ncvb:
NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near {x} = {1.23943}. NIntegrate obtained -2.86229×10^{-16} and 5.515630788991878`*^-17 for the integral and error estimates. >>

-2.86229×10^{-16}

```
Integrate[ Sin[20 x], {x, 0, π / 2} ]
0

NIntegrate[Sin[x] / x, {x, 0, Infinity} ]
1.5708

Pi / 2 // N
1.5708

NIntegrate[1 / Sqrt[x], {x, 0, 1}]
2.

Integrate[1 / Sqrt[x], {x, 0, 1}]
2

NIntegrate[1 / Sqrt[Abs[x]], {x, -1, 2}]
4.82843
```

Singular points of integrands (here $x = 0$) may be indicated in the list:

```
NIntegrate[1 / Sqrt[Abs[x]], {x, -1, 0, 2}]
```

4.82843

Results of integrations should always be checked. For example, by a numerical integration.

```
Clear[a]; f = ArcSinh[a / x]; fn = f /. a -> 1.37 ;
g = Integrate[f, x]
```

$$x\,\mathrm{ArcSinh}\left[\frac{a}{x}\right] + \frac{a\sqrt{a^2+x^2}\,\mathrm{Log}\left[x+\sqrt{a^2+x^2}\right]}{\sqrt{1+\frac{a^2}{x^2}}\,x}$$

```
gn = NIntegrate[fn, {x,1,2}]
```

0.838775

```
gc = (g /. {a -> 1.37, x -> 2}) - (g /. {a -> 1.37, x -> 1})
```

0.838775

Checks of integrations over two or more variables require some care. Compare the formulae below, which shows that 4 terms of the antiderivative (G.: Stammfunktion) are needed.

$$F(x) := \int f(x)\,dx : \qquad \Longrightarrow \qquad \int_{x0}^{x1} f(x)\,dx = F(x1) - F(x0)$$

$$F(x, y) := \int\int f(x, y)\,dy\,dx \qquad \Longrightarrow$$

$$\int_{x0}^{x1}\int_{y0}^{y1} f(x, y)\,dy\,dx \neq F(x1, y1) - F(x0, y0)$$

$$= F(x1, y1) - F(x0, y1) - F(x1, y0) + F(x0, y0)$$

This is shown for the following example, the integral over the function **f** with antiderivative **g.**

```
r = Sqrt[x^2 + y^2 + z^2];

g = x y r / 6 + z^3 ArcTan[x y / (r z)] / 3 + x^3 Log[y + r] / 3 -
    (y^3 + 3 y z^2) Log[x + r] / 6
```

$$\frac{1}{6}x\,y\sqrt{x^2+y^2+z^2} + \frac{1}{3}z^3\,\mathrm{ArcTan}\left[\frac{x\,y}{z\sqrt{x^2+y^2+z^2}}\right] - \frac{1}{6}\left(y^3+3\,y\,z^2\right)\mathrm{Log}\left[x+\sqrt{x^2+y^2+z^2}\right] + \frac{1}{3}x^3\,\mathrm{Log}\left[y+\sqrt{x^2+y^2+z^2}\right]$$

```
f = Simplify[Together[Simplify[D[g, x, y]]]]
```

$$\frac{x^2}{\sqrt{x^2+y^2+z^2}}$$

```
fn = f /. z -> 3.31
```

$$\frac{x^2}{\sqrt{10.9561+x^2+y^2}}$$

```
nn = NIntegrate[fn, {x, .2, .9}, {y, .1, .7}]
```

0.0422782

```
gn = g /. z -> 3.31;
g11 = gn /. {x -> .9, y -> .7} // N ;
g01 = gn /. {x -> .2, y -> .7} // N ;
g10 = gn /. {x -> .9, y -> .1} // N ;
g00 = gn /. {x -> .2, y -> .1} // N ;
dg = g11 - g01 - g10 + g00
```

0.0422782

10.4.3 Options for NIntegrate[]

```
?? NIntegrate
```

NIntegrate[f, {x, x_{min}, x_{max}}] gives a numerical approximation to the integral $\int_{x_{min}}^{x_{max}} f\,dx$.

NIntegrate[f, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}, …]
gives a numerical approximation to the multiple integral $\int_{x_{min}}^{x_{max}} dx \int_{y_{min}}^{y_{max}} dy \ldots f$.

NIntegrate[f, {x, y, …} ∈ reg] integrates over the geometric region reg. ≫

```
Attributes[NIntegrate] = {HoldAll, Protected}

Options[NIntegrate] = {AccuracyGoal → ∞, Compiled → Automatic, EvaluationMonitor → None,
  Exclusions → None, MaxPoints → Automatic, MaxRecursion → Automatic, Method → Automatic,
  MinRecursion → 0, PrecisionGoal → Automatic, WorkingPrecision → MachinePrecision}
```

```
NIntegrate[1 / x, {x, 1, 2}, WorkingPrecision -> 40]
```

```
0.6931471805599453094172321214581765680755
```

```
N[ Log[2], 40 ]
```

```
0.6931471805599453094172321214581765680755
```

10.4.3.1 Numerical integration with high precision

WorkingPrecision, PrecisionGoal and **AccuracyGoal**

WorkingPrecision specifies the number of digits to use for the internal computation.
The accuracy of the results is influenced by **PrecisionGoal** and **AccuracyGoal.**
The default setting **PrecisionGoal -> Automatic**, attemps to get results with a precision equal to **WorkingPrecision** = 10 digits.
In general, **NIntegrate[]** continues until it achieves either of the goals specified by **PrecisionGoal** or **AccuracyGoal**.

```
Clear[x]
g[x_] = x^20 * Exp[x]
```

$e^x x^{20}$

```
Integrate[g[x], {x, 0, 1}] // Expand
```

$-2\,432\,902\,008\,176\,640\,000 + 895\,014\,631\,192\,902\,121\,e$

```
SetAccuracy[ %[[1]], 100]
SetAccuracy[ %%[[2]], 125]
```

```
-2.43290200817664000000000000000000000000000000000000000000000000000000000000\
  000000000000000000000000000000000000000000 × 10^18
```

```
2.4329020081766400001238038307625699486913961699582224511997853081417963754579\
  1658465797553453994566239668736776867393642669698 × 10^18
```

```
inta = SetAccuracy[ % + %%, 100]
```

```
0.1238038307625699486913961699582224511997853081417963754579165846579755345399\
  456623966873677686739364
```

inta is the result of an analytic integration. The numerical evaluation entails the loss of about 18 decimal places. Nevertheless it gives 100 reliable decimal figures.

```
nint = SetAccuracy[ NIntegrate[g[x], {x, 0, 1}], 100]
```

```
0.1238038307625703854375487367178720887750387191772460937500000000000000000000\
  000000000000000000000000
```

```
inta - nint
```

-4.36746152566759649637575253411035449718292083415342024465460054337603312632231326064 × 10^{-16}

The numerical integration above gives only an accuracy of 18 decimal places. But that below with the two options produces a result with 100 correct decimal figures:

```
nint100 = NIntegrate[g[x], {x, 0, 1}, AccuracyGoal → 100, WorkingPrecision → 200]
```

0.123803830762569948691396169958222451199785308141796375457916584657975534539945662396687367768673936426696979316969999534806295254463199654450110908628828101445306953853810809652612201459188629709898 20

```
inta - nint100
```

0. × 10^{-101}

MinRecursion and MaxRecursion

When **NIntegrate[]** tries to do a numerical integral, it samples the integrand at a sequence of points. It it finds that the integrand changes rapidly in a particular region, then it recursively takes more sample points in that region. The parameters **MinRecursion** and **MaxRecursion** specify the minimum and maximum number of levels of recursive subdivions to use. Increasing **MinRecursion** garantees that **NIntegrate[]** will use a larger number of sample points. **MaxRecursion** limits the number of sample points.
SingularityDepth specifies how many levesl of recursive subdivions **NIntegrate[]** will try.

```
Plot[Exp[-x^2], {x, -5, 5}, PlotRange → All]
```




```
N[NIntegrate[Exp[- x^2], {x, -1000, 1000}], 22]
```

1.77245

```
in = SetPrecision[NIntegrate[N[Exp[- x^2], 44], {x, -1000, 1000},
        MinRecursion -> 3, MaxRecursion -> 10], 44]
```

1.7724538509055158819194275565678253769874573

```
SetPrecision[NIntegrate[Exp[- x^2], {x, -1000, 1000}], 44]
```

1.7724538509055154378302177065052092075347900

```
% - %%
```

-4.4408920985006261616945266672 × 10^{-16}

```
Integrate[Exp[-x^2], {x, -1000, 1000}]
```

$\sqrt{\pi}$ Erf[1000]

```
ia = N[%, 44]
```

1.7724538509055160272981674833411451827975495

```
N[√π Erf[Infinity], 44]
```

1.7724538509055160272981674833411451827975495

```
in - ia
```

$-1.4537873992677331980581009 22\times 10^{-16}$

In order to check the accuracy of the analytic result, an asymptotic formula for large a is obtained by partial integrations:

$$\int_0^a e^{-x^2}\,dx = \int_0^\infty e^{-x^2}\,dx - \int_a^\infty e^{-x^2}\,dx = \sqrt{\pi}/2 - \int_a^\infty \underbrace{(-2x)\,e^{-x^2}}_{u'} \underbrace{\left(\frac{-1}{2x}\right)}_{v}\,dx$$

$$= \sqrt{\pi}/2 - \left[\frac{e^{-x^2}}{2x}\right]_a^\infty + \int_a^\infty e^{-x^2}\left(\frac{1}{2x^2}\right)dx = \sqrt{\pi}/2 - 0 + \frac{e^{-a^2}}{2a} + \ldots$$

This procedure could be continued, but this is not necessary if a is sufficiently large as it is here:

```
N[√π, 44]
```

1.7724538509055160272981674833411451827975495

$$\frac{e^{-a^2}}{2a} \,/.\, a \to 100 \,//\, N$$

$5.677419326573681\times 10^{-4346}$

10.4.3.2 Integration of a discontinuous function

```
Clear[f, x]
f[x_] := Which[ x < 0, 0, x < 1, 1, x >= 2, 2 - x, x >= 1, x];
```

```
Plot[f[x], {x, -1, 4}, PlotStyle → Thickness[0.01`]]
```




```
NIntegrate[f[x], {x, -1, 4}]
```

0.5

10.4.3.3 Numeric Principal Value Integrals

The definition of the principal value given in 10.4.1.7 and the methods to evaluate it apply also here.

Evaluation by the option **Method -> "PrincipalValue"**

```
Integrate[Sin[x] / x^2, {x, -1, 2}]
```

Integrate::idiv: Integral of $\frac{\text{Sin}[x]}{x^2}$ does not converge on {-1, 2}. ≫

$$\int_{-1}^{2} \frac{\text{Sin}[x]}{x^2} \, dx$$

```
Integrate[Sin[x] / x^2, {x, -1, 2}, PrincipalValue -> True]
```

$$-\text{CosIntegral}[1] + \text{CosIntegral}[2] + \text{Sin}[1] - \frac{\text{Sin}[2]}{2}$$

```
N[%] // Chop
```

0.472399

```
NIntegrate[Sin[x] / x^2, {x, -1, 0, 1, 2}, Method -> "PrincipalValue"]
```

NIntegrate::izero:
Integral and error estimates are 0 on all integration subregions. Try increasing the value of the MinRecursion option. If value of integral may be 0, specify a finite value for the AccuracyGoal option. ≫

0.472399

$$\text{Integrate}\left[\frac{1}{x - x^2}, \{x, -1, 2\}, \text{PrincipalValue} \to \text{True}\right]$$

Log[4]

```
N[%]
```

1.38629

$$\text{NIntegrate}\left[\frac{1}{x - x^2}, \{x, -1, 0, 1, 2\}, \text{Method} \to \text{"PrincipalValue"}\right]$$

1.38629

Note: The very last integral above has two first oder poles, one at $x = 0$, the other at $x = 1$.
It is essential that in the argument list of **NIntegrate[]** the position(s) of the first order pole(s) is (are) given accurately. Otherwise the command will yield a wrong result. So the position of the pole must be found beforehand to high precision !

```
n[x_] = 1 + x^2;
d[x_] = x^5 + x^2 + x + 0.5;
```

$$\int_{-1}^{1} n[x] / d[x] \, dx$$

Integrate::idiv: Integral of $\frac{1}{0.5 + x + x^2 + x^5} + \frac{x^2}{0.5 + x + x^2 + x^5}$ does not converge on {-1, 1}. ≫

$$\int_{-1}^{1} \frac{1 + x^2}{0.5 + x + x^2 + x^5} \, dx$$

```
Plot[n[x] / d[x], {x, -2, 1}, AxesLabel → {"x", "n(x)/d(x)"}]
```




```
so = Solve[d[x] == 0]
```

```
{{x → -0.808309}, {x → -0.389139 - 0.520029 i}, {x → -0.389139 + 0.520029 i},
 {x → 0.793293 - 0.914874 i}, {x → 0.793293 + 0.914874 i}}
```

```
xp = x /. so[[1]]
```

```
-0.808309
```

```
Integrate[n[x] / d[x] , {x, -1, 1}, PrincipalValue -> True ] // Chop
```

```
4.3803
```

That the above analytical principal value integration works is an exception.
Exercise 10.10 is an example to the contrary.

```
inPV = NIntegrate[n[x] / d[x] , {x, -1, xp, 1}, Method -> "PrincipalValue"]
```

```
4.3803
```

```
NIntegrate[n[x] / d[x] , {x, -1, -0.78, 1}, Method -> "PrincipalValue"]
```

```
NIntegrate::ncvb :
  NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near {x} = {0.0283577}. NIntegrate
     obtained -54.309 and 54.82013079799068 for the integral and error estimates. ≫
```

```
-50.9524
```

An inaccurate value for the pole position leads to a wrong result !

Evaluation by complex integration

For this method no accurate value of the pole position is needed. But the positions of all poles must be known. The integration paths must have the same starting and end point as the real original path; in addition, the resulting closed curve CU + CL must not include any other singularity than the poles on the original path.

```
plro = ListPlot[{Re[x], Im[x]} /. so];
```

```
 CU = {{-1, 0}, {0, 0.5}, {1, 0}};
CL = {{-1, 0}, {0, -0.5}, {1, 0}};
plpa = ListLinePlot[{CU, CL}];
```

```
Show[plro, plpa, PlotRange → All, Epilog →
  {Text[Subscript["C", "L"], {0.2, -0.5}], Text[Subscript["C", "U"], {0.2, 0.5}]}]
```




```
iU = NIntegrate[n[x] / d[x] , {x, -1, 0.5 I, 1}]
```

4.3803 − 3.42218 ⅈ

```
iL = NIntegrate[n[x] / d[x] , {x, -1, -0.5 I, 1}]
```

4.3803 + 3.42218 ⅈ

```
ih = (iU + iL) / 2 // Chop
```

4.3803

```
ih - inPV
```

4.70113×10^{-12}

10.5 Integral Transforms

10.5.1 Laplace Transform

The Laplace Transform of a function f(t) is defined as

$$F(s) := \int_0^\infty f(t)\, e^{-s\,t}\, dt := \mathcal{L}(f(t)) \qquad (1)$$

LaplaceTransform[*expr*, *t*, *s*]	gives the Laplace transform of ***expr***

Laplace Transform of a constant c: $\mathcal{L}$(c) =

```
LaplaceTransform[c, t, s]
```

$\frac{c}{s}$

```
LaplaceTransform[t^n, t, s]
```

$s^{-1-n}\,\text{Gamma}[1+n]$

```
LaplaceTransform[Sin[ω t], t, s]
```

$\frac{\omega}{s^2+\omega^2}$

```
LaplaceTransform[Cos[ω t], t, s]
```

$\frac{s}{s^2+\omega^2}$

```
LaplaceTransform[t Sin[ω t], t, s]
```

$\frac{2\,s\,\omega}{(s^2+\omega^2)^2}$

```
LaplaceTransform[t Cos[ω t], t, s]
```

$\frac{s^2-\omega^2}{(s^2+\omega^2)^2}$

The Heaviside unit step function is a discontinuous function.

```
ft1 = If[t ≤ a, 0, 1];
sua = a → 1;
Plot[ft1 /.sua, {t, 0, 4}, AspectRatio → Automatic, AxesLabel → {"t", "f(t)"},
 Ticks → {{0, {a, "a"}} /.sua, {0, 1}}, PlotStyle → Thick]
```


```
LaplaceTransform[ft1, t, s]
```

$$\begin{cases} \frac{1}{s} & a \le 0 \\ \frac{e^{-a\,s}}{s} & \text{True} \end{cases}$$

```
Integrate[Exp[- s t], {t, a, Infinity}]
```

$\text{ConditionalExpression}\left[\frac{e^{-a\,s}}{s},\ \text{Re}[s] > 0\right]$

A pulse lasting from a to b is shown in the following drawing:




```
Integrate[Exp[- s t], {t, a, b}]
```

$$\frac{e^{-a\,s} - e^{-b\,s}}{s}$$

```
LaplaceTransform[BesselJ[0, a t], t, s]
```

$$\frac{1}{\sqrt{a^2 + s^2}}$$

```
LaplaceTransform[BesselJ[n, a t], t, s]
```

$$\frac{a^n \left(s + \sqrt{a^2 + s^2}\right)^{-n}}{\sqrt{a^2 + s^2}}$$

```
% /. n → 3 // Simplify
```

$$\frac{a^3}{\sqrt{a^2 + s^2}\left(s + \sqrt{a^2 + s^2}\right)^3}$$

```
Apart[%]
```

$$\frac{1}{a} + \frac{4\,s^2}{a^3} - \frac{a}{s\sqrt{a^2 + s^2}} + \frac{\sqrt{a^2 + s^2}}{a\,s} - \frac{4\,s\sqrt{a^2 + s^2}}{a^3}$$

10.5.1.1 Transformation of Differential Equations

Deriving the defining integral $F(s) = \int_0^\infty f(t)\, e^{-s\,t}\, dt$, w.r.t. t gives :

$$\mathcal{L}(f'(t)) = s\,F(s) - f(+0),$$

$$\mathcal{L}(f''(t)) = s^2\,F(s) - s\,f(+0) - f'(+0), \quad \ldots$$

Applying the Laplace transform changes the following linear differential equation with given c1, c2 and f (t) :

$$\ddot{x}(t) + c1\,\dot{x}(t) + c2\,x(t) = f(t)$$

into a linear algebraic equation :

$$\left[s^2\,X(s) - s\,x(+0) - x'(+0)\right] + c1\,[s\,X(s) - f(+0)] + C0 = F(s).$$

This may be solved for the unknown amplitude
$X(s)$ and $x(t)$ is found by the inverse Laplace transform .

10.5.1.2 The Inverse Laplace Transform

The Inverse Laplace Transform is : $\mathcal{L}^{-1}(F(s)) := \frac{1}{2\pi i}\int_{c-i\infty}^{c+i\infty} F(s)\, e^{s\,t}\, ds$

where c is an arbitrary positive constant chosen such that the contour of integration lies to the right of all singularities in $F(s)$.

InverseLaplaceTransform[*expr*, *s*, *t*]	gives the inverse Laplace transform of *expr*

```
InverseLaplaceTransform[s/(s^2 + ω^2), s, t]
```

Cos[t ω]

```
InverseLaplaceTransform[c / s, s, t]
```

c

```
InverseLaplaceTransform[e^(-a s)/s, s, t]
```

HeavisideTheta[-a + t]

```
InverseLaplaceTransform[c, s, t]
```

c DiracDelta[t]

10.5.1.3 A Voltage Jump applied to a Series Circuit

A series circuit consists of a condensor of capacity C (charge q = $\int$ i dt = C V) and a coil having an Ohmic resistance R (V = R i) and an inductivity L (V = L di/dt.). A Voltage step V(t) = V0 θ(t) is applied to this circuit. The current is represented as a Laplace transform F(s) = $\mathcal{L}$(i(t)).

$$\int i\,dt / C + R i + L\,di/dt = V(t), \quad i/C + R\,di/dt + L\,d^2 i/dt^2 = V0\,d\theta/dt = V0\,\delta(t);$$

$$F / C + R s F + L s^2 F == V0 .$$

```
fi = V0 / (s R + s^2 L + C)
```

$$\frac{V0}{C + R s + L s^2}$$

```
ti = InverseLaplaceTransform[fi, s, t]
```

$$-\frac{\left(e^{\left(-\frac{R}{2L}-\frac{\sqrt{-4 C L+R^2}}{2L}\right) t} - e^{\left(-\frac{R}{2L}+\frac{\sqrt{-4 C L+R^2}}{2L}\right) t}\right) V0}{\sqrt{-4 C L + R^2}}$$

```
svd = {V0 → 10, R → 22, L → 110, C → 1};
(ti /. svd // N) /. 2.718281828459045` → e
```

$$-1.50756 \left(e^{-0.130151\,t} - 1.\,e^{-0.0698489\,t}\right)$$

```
Plot[Evaluate[ti /.svd], {t, 0, 57}, PlotRange → All, AxesLabel → {"t", "I(t)"}]
```




```
svs = {V0 → 10, R → 22, L → 110, C → 19};
```

```
(ti /. svs // N // Expand // Chop) /. 2.718281828459045` → ⅇ
```

$(0. + 0.11268\, i)\, e^{(-0.1-0.403395\, i)\, t} - (0. + 0.11268\, i)\, e^{(-0.1+0.403395\, i)\, t}$

```
Plot[Evaluate[ti /.svs], {t, 0, 57}, PlotRange → All, AxesLabel → {"t", "I(t)"}]
```



10.5.2 Fourier Transform

`FourierTransform[f (t), t, ω]`	$F(\omega) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} f(t)\, e^{i\omega t}\, dt$
`InverseFourierTransform[F (ω), t, ω]`	$f(t) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} F(\omega)\, e^{-i\omega t}\, d\omega$
`FourierCosTransform[f (t), t, ω]`	$F(\omega) = \sqrt{\frac{2}{\pi}} \int_0^{\infty} f(t) \cos(\omega t)\, dt$
`InverseFourierCosTransform[F (ω), ω, t]`	$f(t) = \sqrt{\frac{2}{\pi}} \int_0^{\infty} F(\omega) \cos(\omega t)\, d\omega$
`FourierSinTransform[f (t), t, ω]`	$F(\omega) = \sqrt{\frac{2}{\pi}} \int_0^{\infty} f(t) \sin(\omega t)\, dt$
`InverseFourierSinTransform[F (ω), ω, t]`	$f(t) = \sqrt{\frac{2}{\pi}} \int_0^{\infty} F(\omega) \sin(\omega t)\, d\omega$

```
ft = 1/(t^2 + a^2);
fw = FourierTransform[ft, t, ω]
```

$\frac{1}{2a} e^{-a\omega} \sqrt{\frac{\pi}{2}} \Big((-1 + e^{2a\omega})\, \text{Sign}[\omega]\, (-1 + \text{Sign}[\text{Abs}[\text{Re}[a]]]) +$
$2\, (e^{2a\omega}\, \text{HeavisideTheta}[-\omega\, \text{Sign}[\text{Re}[a]]] + \text{HeavisideTheta}[\omega\, \text{Sign}[\text{Re}[a]]])$
$\text{Sign}[\text{Re}[a]] \Big)$

```
PowerExpand[%]
```

$\frac{1}{2a} e^{-a\omega} \sqrt{\frac{\pi}{2}} \Big((-1 + e^{2a\omega})\, \text{Sign}[\omega]\, (-1 + \text{Sign}[\text{Abs}[\text{Re}[a]]]) +$
$2\, (e^{2a\omega}\, \text{HeavisideTheta}[-\omega\, \text{Sign}[\text{Re}[a]]] + \text{HeavisideTheta}[\omega\, \text{Sign}[\text{Re}[a]]])$
$\text{Sign}[\text{Re}[a]] \Big)$

```
fm = 1/Sqrt[2 π] Integrate[ft Exp[I ω t],
   {t, -Infinity, Infinity}, Assumptions -> {a > 0 && Element[ω, Reals]}]
```

$$\frac{e^{-a\,\text{Abs}[\omega]}\sqrt{\frac{\pi}{2}}}{a}$$

```
FourierCosTransform[ft, t, ω]
```

$$\sqrt{\frac{1}{a^2}}\, e^{-\frac{\omega}{\sqrt{\frac{1}{a^2}}}}\sqrt{\frac{\pi}{2}}$$

```
PowerExpand[%]
```

$$\frac{e^{-a\,\omega}\sqrt{\frac{\pi}{2}}}{a}$$

```
Sqrt[2/π] Integrate[ft Cos[ω t], {t, 0, Infinity}, Assumptions -> {a > 0 && ω > 0}]
```

$$\frac{e^{-a\,\omega}\sqrt{\frac{\pi}{2}}}{a}$$

```
fst = (UnitStep[t] + UnitStep[a - t] - 1)
```

-1 + UnitStep[a - t] + UnitStep[t]

```
sa = {a → 1};
```

```
Plot[fst /.sa, {t, -1, 2}, PlotRange → {0, 1.1`}, PlotStyle → Thickness[0.009`],
 Epilog → Text["0", {0, -0.065`}], ImageSize → 250,
 PlotLabel → "f(t) = -1+UnitStep[1-t]+UnitStep[t]\n", AxesLabel → {"t", "f(t)"}]
```




```
fw = FourierTransform[ft, t, ω]
```

$$\frac{1}{2\,a} e^{-a\,\omega}\sqrt{\frac{\pi}{2}}\left(\left(-1 + e^{2\,a\,\omega}\right)\text{Sign}[\omega]\left(-1 + \text{Sign}[\text{Abs}[\text{Re}[a]]]\right) + 2\left(e^{2\,a\,\omega}\,\text{HeavisideTheta}[-\omega\,\text{Sign}[\text{Re}[a]]] + \text{HeavisideTheta}[\omega\,\text{Sign}[\text{Re}[a]]]\right)\text{Sign}[\text{Re}[a]]\right)$$

```
Simplify[fw , a > 0]
```

$$\frac{e^{-a\,\omega}\sqrt{\frac{\pi}{2}}\left(e^{2\,a\,\omega}\,\text{HeavisideTheta}[-\omega] + \text{HeavisideTheta}[\omega]\right)}{a}$$

```
fm = 1/Sqrt[2 π] Integrate[fst Exp[I ω t],
   {t, -Infinity, Infinity}, Assumptions -> {a > 0 && Element[ω, Reals]}]
```

$$-\frac{i\left(-1 + e^{i\,a\,\omega}\right)}{\sqrt{2\,\pi}\;\omega}$$

```
fw = InverseFourierTransform[fm, ω, t]
```

$$\frac{1}{2}\left(\text{Sign}[a-t]+\text{Sign}[t]\right)$$

```
Plot[fw /.sa, {t, -1, 2}, PlotRange → {0, 1.1}, PlotStyle → Thick, ImageSize → 250,
 PlotLabel → "f(t) = [Sign(a - t) + Sign(t)]/2\n", AxesLabel → {"t", "f(t)"}]
```

f(t) = [Sign(a – t) + Sign(t)]/2



But this integral may be evaluated easily by Cauchy's residue theorem.

```
fss = (UnitStep[a + t] + UnitStep[a - t] - 1)
```

$$-1+\text{UnitStep}[a-t]+\text{UnitStep}[a+t]$$

```
Plot[fss /.sa, {t, -2, 2}, PlotRange → {0, 1.1`}, PlotStyle → Thick,
 PlotLabel → "f(t) = -1 + UnitStep[1-t] + UnitStep[1 + t]\n",
 AxesLabel → {"t", "f(t)"}, ImageSize → 250]
```

f(t) = –1 + UnitStep[1–t] + UnitStep[1 + t]




```
fs = FourierCosTransform[fss, t, ω]
```

$$\frac{\sqrt{\frac{2}{\pi}}\ \text{Sin}[a\,\omega]\ \left(1+\text{HeavisideTheta}[a]-\text{UnitStep}[a]\right)}{\omega}$$

```
InverseFourierTransform[fs, ω, t]
```

$$\frac{1}{2}\left(\text{Sign}[a-t]+\text{Sign}[a+t]\right)$$

```
Plot[% /.sa, {t, -2, 2}, PlotRange → {0, 1.1},
  PlotStyle → Thick, AxesLabel → {"t", "f(t)"}, ImageSize → 250];
```

Gives the same picture as above.

```
fi = InverseFourierCosTransform[fs, ω, t]
```

$$\frac{1}{2}\left(\frac{a-t}{\sqrt{(a-t)^2}}+\frac{a+t}{\sqrt{(a+t)^2}}\right)$$

```
Simplify[fi, {a > 0 && Element[t, Reals]}]
```

$$\begin{cases} 1 & a \geq t \,\&\&\, a + t \geq 0 \\ 0 & \text{True} \end{cases}$$

```
Plot[Chop[fi /.sa], {t, -2, 2}];
```

Gives the same picture as above.

```
fu = -t UnitStep[-a + t] + t UnitStep[a + t]
```

```
-t UnitStep[-a + t] + t UnitStep[a + t]
```

```
Plot[fu /.sa, {t, -2, 2}, PlotRange → All,
 AxesLabel → {"t", "fu(t)"}, PlotStyle → Thick, ImageSize → 250]
```




```
fut = FourierSinTransform[fu, t, ω]
```

$$\frac{\sqrt{\frac{2}{\pi}}\,(\text{HeavisideTheta}[-a] - \text{HeavisideTheta}[a])\,(\omega\,\text{Abs}[a]\,\text{Cos}[a\,\omega] - \text{Sin}[\omega\,\text{Abs}[a]])}{\omega^2}$$

```
Plot[fut /.sa, {ω, -15, 15}, AxesLabel → {"ω", "Fu(ω)"},
 PlotStyle → Thick, ImageSize → 300]
```



10.6 Exercises

10.1. Calculate the following limits (obligatory are 1) to 3)):

1) $[\sin(x) - x \cos(x)]/ x^n$ for $x \to 0$, $n = 1,2,3,4$;
2) $\ln(x) /(1 - x)$ for $x \to 1$;
3) $[\sqrt{(x^2 - 3)} - x]\, x^n$ for $x \to \infty$, $n = 0, 1$;
4) $J_n(x)/x^n$ for $x \to 0$, $n = 1,2$, $n \in IR_+$;
5) $Y_n(x)\, x^n$ for $x \to 0$, $n = 1,2$, $n \in IR_+$;

$J_n(x)$ = BesselJ[n,x] Besselfunction of first kind of order n;
$Y_n(x)$ = BesselY[n,x] Besselfunction of second kind of order n.

10.2. Expand the following functions into Taylor's or Laurent's series around x = 0:
1) $\sin^2(x)$, 2) $\sin(\sqrt{x})$, 3) $\sin^{-2}(x)$, 4) e^{-x2} , 5) $\text{ctg}^3(x)$, 5) $\sin(\sqrt{(a x + b)})$.

10.3. Calculate the following residues (2) is obligatory) :
at $x = 0$: 1) e^{-x2}/x^2 2) e^{-i4x}/x^3 3) $J_n(x)/x^{n+1}$, $n \in IN$;
at $x = 2\pi$: 4) $(x^2 + 3 x - 1)/\sin^2 x$

10.4. Do the integrals given below analytically; check the result by differentiating it and by numeric integration (2) is obligatory) :
1) $\int dx\, (3 x + 1)/[(x + 1)^2 (x - 2)]$, 2) $\int dx\, \sin(5 x) \cos 3(x)$,

3) $\int$ dx $\int$dy x y/r^2, r = $\sqrt{(x^2 + y^2 + z^2)}$, 4) $\int$ dx $\int$dy x y/r ;

10.5. Compute for the octant (x ≥ 0, y ≥ 0, z ≥ 0) of the ellipsoid

$$(x/a)^2 + (y/b)^2 + (z/c)^2 = 1$$

the tensor of the moment of inertia w.r.t. the following points:
1) x = y = 0, 2) the center of gravity, 3) x = a/2, y = b/3, z = 4c/9.
Compute the eigenvalues and eigenvectors of these tensors for
4) a = b, 5) a = 2, b = 5, c = 4.

10.6 Approximate the sine in the intervall (0, π) by a 3 rd degree polynomial. The coefficients should be determined by the values of the sine at the points (0, π/2, π) and by the condition that the polynomial should have a horizontal tangent at π/2. Plot these two curves. Compare the area between this polynomial and the x-axis with the area under the sine curve.

10.7. The basis, a, and the length of the arc, b, of the segment of a circle are given. Find an approximate analytic expression for the central angle α (0 ≤ α < 180°) and the height, h, of the segment as functions of a and b for 0 ≤ a/b ≤ 1. Compare the approximate analytic results to accurate numeric results in tables and graphs. Hint: Half of the central angle is determined by a transcendental equation, for which an approximate analytic result is obtained by one Newton iteration. The height can be calculated with this angle.

10.8. Calculate the following integrals by expanding the integrands around t = 0 and integrating the resulting series (10 terms) term by term. Check the accuracy of this approximate result with that obtained by numeric integration.
1) $\int_0^z$ dt/ $(1 - t^4)^{1/2}$, 0 ≤ z ≤ 0.8 .
2) $\int_0^z$ dt $t^{-1/2}$ $(t - 1/2)^{-1/2}$ $(1 - t + t2/2)^{-1/2}$, 0 ≤ z ≤ 0.2.

10.9+ Evaluate the integral $\int_0^1 dx\ x^{20}\ e^x$ in 3 ways:
1) N[Integrate], 2) NIntegrate[ ] ,
3) Series expansion and subsequent term by term integration.
Explain the differences in the results and devise all methods such that each gives a result correct to 10 decimal places.

10.10 Investigate the following integral and give its true value or limit.

$$\int_{-1}^{1} \frac{x-1}{x^7 + x^3 + 1} \, dx$$

10.11 Find all zeros of the transcendental equation f(z) = 1/2 z e^z - 1 = 0 contained within the square 50 {1+i,-1+i,-1-i,-1-i}. Show that you have found all zeros.
Hint: The large range of the square renders an accurate plot of |f(z)| impractical.
It is more advantageous to use the theorem: $\int_C$ dz f'(z)/f(z)/(2π I) = N - P, where N (P) is the number of zeros (poles) located within the counture C. The value of the integral may be found by numeric integration.

10.12 Evaluate the following integral analytically and numerically.

In[6]:=

$$\int_1^2 dx \int_1^{x^2} dy\ (x^2 + y^2)$$

10.13 Find a method to systematically replace the limit

```
(f[y]-f[x])/(y-x) /. y->x
```

by f'[x], no matter what the function f is.

10.14 Define a function which calculates the Wronskian of n functions. Apply it to {sin x, sin 2x, sin 3x} and find the simplest expression for the result.

10.15 Compute the indefinite integral: $\int |x| \, dx$.

10.16 Calculate the Laplace transform of the function f[t], which is zero outside the interval [0 < a, b] and a triangle with the top ($\frac{a}{2} + \frac{b}{2}$,1) within this intervall.

10.17 Laplace transform: Find the images of the following functions:
1) $t^a e^{bt}$, 2) (1 - e^{-t})/t , 3) ln t , 4) $e^{-t^2/4}$,
5) cos(x $\sqrt{t}$)/$\sqrt{t}$, 6) sin(x $\sqrt{t}$) , 7) ch(x $\sqrt{t}$)/$\sqrt{t}$, 8) sh(x $\sqrt{t}$) ,
9) $e^{-x^2/4t}/\sqrt{t}$, 10) $e^{-x^2/4t}/\sqrt{t^3}$.

10.18 Laplace transform: Find the preimages (= Urbilder) of the following functions:

1) $s^n/(s^3 + \omega^3)$, n = 0, 1, 2; 2) $s^n/(s^4 + 4\,\omega^4)$, n = 0, 1, 2, 3;

3) $s^n/(s^4 - \omega^4)$, n = 0, 1, 2, 3; 4) Exp[s^2], 5) Exp[- a $\sqrt{s}$]/s .

10.19 The two branches of a parallel circuit consist of 1) a coil (Ohmic resistance R, inductivity L)
2) of a condensor (capacity C) preceeded by a small resistance r.
A voltage step of magnitude V0 is applied to the circuit. Compute and plot the total current and that in each branch.

10.20 Fourier transform: Perform the Fourier transform for the following function:

$$\cos(\omega 0\ t).$$

10.21 Find den following limit: $\lim_{n\to\infty}(-1)^n \binom{-1/2}{n}\sqrt{\pi n}$.

Hint: $(-1)^n \binom{-1/2}{n} = \frac{(-1/2)\,(-3/2)\ldots(-(2n-1)/2)}{n!}(-1)^n = \frac{(1/2)\,(3/2)\ldots(n-1/2)}{n!}$